

681
A 45

681.3


КОМПЬЮТЕР САВОДХОНЛИГИ

С.А. Арипов, Ш.Р. Юсупов, И.Р. Камолов

БЕЙСИК
дастурлаш тили

19. 01. 2010

201014 - 24



Тошкент - "Наврӯз"

"Компьютер саводхонлиги" тўпламининг мазкур учинчи китобида Бейсик дастурлаш тилининг "Правец-8", "Агат" ва "Тошкент" компьютерларида ишлашга мўлжалланган версияси ёритилди. Асосий операторлар тавсифи ва улардан фойдаланиш йўллари мисоллар воситасида тушунтирилган. Қўлланма олий ўқув юртлари, ўрта махсус ўқув юртлари, хунар-техника билим юртлари, ўрта умум таълим мактаблари ўқитувчи ҳамда ўқувчилари учун мўлжалланган.

КОМПЬЮТЕР САВОДХОНЛИГИ

Тўплам. 3- китоб.

Арипов Салоҳиддин Алимович
Юсупов Шовкатжон Рихсибоевич
Камолов Илҳом

Бейсик дастурлаш тили.

Муҳаррир: *Т.А. Нурмухамедов.*
Техник муҳаррир: *Т.А. Нурмухамедов, Т.В. Большакова*
Рассом: *С. Саидрасулов*
Мусаҳҳиҳа: *М.А. Нурмухамедова*

Китобнинг оригинал макети Ventura Publisher нашр системаси ёрдамида тайёрланди. Dutch гарнитураси. Босмахонага топширилди 03.10.94 йил. Босишга рухсат этилди 19.10.94 йил. Қоғоз бичими 84x108 1/32. Офсет усулида босилди. Шартли б.т. 3,36. Шартли кр.-отт. 3,60. Нашр б.т. 3,25. 50000 нуска. Буюртма №133.

"Наврўз" нашриёти, 700000. Тошкент, Мовароуннаҳр кўчаси, 4. Шартнома №25

Тошкент картография фабрикасида босилди. Полковник Асом Муҳитдинов кўчаси 6-уй. Баҳоси шартнома асосида.

© "Шуъла" маркази, "Кўҳи-Нур" фирмаси, 1994

Сўз боши

Бейсик дастурлаш тили нисбатан соддалиги, компьютерга қўйиладиган талабларнинг озлиги туфайли ўқувчиларни, талабаларни ҳисоблаш техникаси асослари билан таништиришда кенг қўлланиб келинмоқда.

Бейсик тилининг турли шахсий компьютер имкониятларини ҳисобга олган ўндан ортиқ лаҳжаси (версияси) мавжуддир. Республикамизнинг аксарият ўқув муассасалари "Правец-8", "Агат", "Тошкент" шахсий компьютерлари билан таъминланганлигини ва Бейсик тилининг мазкур лаҳжасига бағишланган қўлланмалар озлиги, жуда кам нусхада чоп этилганлигини ҳисобга олган ҳолда, эътиборингизга мазкур қўлланмани ҳавола этдик.

Қўлланма олти бўлимдан иборат бўлиб, унда Бейсик тилининг "Правец-8", "Агат", "Тошкент" шахсий компьютерларига мос лаҳжаси ҳақида бошланғич маълумотлар, асосий буйруқлар ва операторлар тавсифи ва улардан дастур тузишда фойдаланиш йўллари мисоллар воситасида ёритилган.

Китоб ҳажми чегаралангани, "Правец-8" ва "Агат", "Тошкент" компьютерларининг график имкониятлари бирмунча фарқ қилишини ҳисобга олган ҳолда Бейсик тилининг график операторлари тавсифи берилмади. "Правец-8" компьютерининг график имкониятлари нашрга тайёрланаётган "Танишинг: Правец-8" китобида ёритилади.

Муаллифлар қўлланма, унда мавжуд бўлиши мумкин бўлган камчиликлар тўғрисидаги фикр ва мулоҳазаларингизни мамнуният билан қабул қиладилар.

1. Асосий тушунчалар.

Бейсик тили юқори даражали дастурлаш тиллари синфига мансуб тилдир. "Бейсик" (BASIC) сўзи қуйидаги инглиз сўзларининг бош ҳарфларидан тузилган: Beginner's All-purpose Symbolic Instruction Code (BASIC). Буни сўзма-сўз таржима қиладиган бўлсак "Бошловчилар учун белгили буйруқлардан тузилган кўп мақсадли тил (код)" деганидир. Бейсик тили ва унда тузилган биринчи дастур 1964 йилда Дартмут (АҚШ) коллежининг хизматчилари Джон Кемени ва Томас Куртц томонидан яратилган. Уларнинг асосий мақсади мутахассис бўлмаган фойдаланувчиларнинг ЭҲМ билан мулоқотини, ишини енгиллаштириш эди. Бейсик тили соддалиги ва компьютер хотирасига қўйиладиган талабларнинг жуда озлиги сабабли бошқа юқори даражали дастурлаш тиллари қатори бутун дунёда ШЭҲМлар учун кенг қўлланиладиган тил бўлиб қолди.

1.1. Бейсик тилининг асосий элементлари.

Бейсикда белгиларнинг тўплами лотин алифбосининг катта ва кичик ҳарфлари, сонлар, махсус белгилардан иборат. Сонлар араб рақамларидан (0 дан 9гача) тузилади. Булардан ташқари, тилнинг баъзи операторлари рус алифбоси ҳарфларидан фойдаланишга имкон яратади.

Арифметик ва таққослаш амалларининг Бейсик тилида ёзилиши:

Оддий ёзилиши	Бейсикда ёзилиши
x^n	$x^{\wedge}n$
$a \cdot b$	$a*b$
$a : b$	a/b
$a + b$	$a+b$
$a - b$	$a-b$
$a = b$	$a=b$
$a < b$	$a<b$
$a \leq b$	$a<= b$
$a > b$	$a > b$
$a \geq b$	$a >= b$
$a \neq b$	$a <> b$

Булардан ташқари, бейсик тилида мантиқий амаллар сифатида OR (ёки), AND (ва), NOT (эмас) каби ёрдамчи сўзлар ишлатилади.

1.2. Берилганларни ифодалаш.

Бейсик тилида катталиклар иккига бўлинади: ўзгарувчи ва ўзгармас (константа) миқдорлар. Дастур бажарилиш жараёнида қиймати ўзгарадиган катталиклар ўзгарувчилар деб, қиймати ўзгармайдиган катталиклар ўзгармаслар деб аталади. Ҳар бир ўзгарувчи ўзининг мустақил номига эга, у идентификатор деб аталади. Ўзгарувчининг номи ҳар доим лотин алифбосининг ҳарфларидан бошланиши шарт, керак бўлиб қолган тақдирда ундан сўнг араб рақамларини ёзиб ишлатиш мумкин. Ўнли ва каср сонларни, русча ҳарфларни ўзгарувчининг номи сифатида ишлатиш мумкин эмас. Масалан: C5, SD, MIN, PROG - ишлатиш мумкин бўлган ўзгарувчиларнинг номи; Физика, C55.3, PR-L - ишлатиш мумкин бўлмаган ўзгарувчиларнинг номи. Ўзгарувчиларнинг номларини қўйишда қўйидаги қоидаларга амал қилинади:

1. Ўзгарувчилар номи 6 та белгидан ошмаслиги керак. Ўзгарувчининг номлари биринчи икки белгилари билан фарқланадилар. Масалан ALFA, ALLA, AL8, AL ўзгарувчилари компьютер учун бир ўзгарувчини билдиради — AL.

2. Ўзгарувчининг номи лотин ҳарфидан бошланиши шарт. Лотин ҳарфидан сўнг араб рақамларини ишлатиш мумкин. Рус ҳарфларини ишлатиш мумкин эмас.

3. Ўзгарувчининг номи компьютернинг махсус сўзлари билан устма-уст тушмаслиги керак, масалан, ўзгарувчиларнинг номлари ORE, OR, ORDER, OR2 ва шу каби сўзлар бўлиши мумкин эмас, чунки OR сўзи компьютернинг махсус сўзидир.

Бейсикда ўзгарувчилар 3 хил бўлади: ҳақиқий, бутун ва ҳарфий. Ҳақиқий сонлар Бейсик тилида математикада қабул қилинган оддий ўнли каср кўринишида ёзилади, фақат соннинг бутун қисмини каср қисмидан ажратувчи вергул ўрнига нуқта қўйилади. Масалан: 0.35; -96.105; 4.006

Жуда катта ва жуда кичик сонларни ёзишда уларни ўннинг даражаси сифатида (стандарт кўринишида) ифодалаш мумкин. Бунинг учун E белгиси ишлатилади ва улардан кейин мусбат ёки манфий ишорали бутун сон ёзилади. Агар сон мусбат бўлса "+" белгисини қўйиш шарт эмас. Масалан: ёруғликнинг бўшлиқдаги тезлиги $V=300000$ км/сек (ёки $3 \cdot 10^5$ км/сек) ни

ўннинг даражаси сифатида $3E5$ деб ёзиш мумкин, конденсатор сифимининг қиймати $C=10 \text{ Мкф}$ ($10 \cdot 10^{-6} \text{ Ф}$)ни эса $10E-6$ каби ёзилади.

Сонларни кўрсаткичли шаклда ёзиш қоидалари:

1. Е белгисининг олдида ихтиёрий сон бўлиши мумкин;
2. Е белгисидан кейин эса фақат бутун сон ёзилади.

Бутун сонларнинг Бейсикдаги ёзилиши уларнинг оддий ёзувдаги кўринишидан фарқ қилмайди ("+" белгисини ёзмаслик ҳам мумкин): -709, +510, 510, 12, -36. Бутун сонларни алоҳида таъкидлаш учун унинг номига % (процент) белгиси қўшилади. Масалан: А%, MIN%, В7%.

Ҳарфий катталиклар деб қўштирноқ ичига олинган ЭҲМ нинг ихтиёрий белгилари кетма-кетлигига айтилади. Масалан: "БЕЙСИК", "Функциянинг аниқланиш соҳаси", "КЕЛЬВИН", "1234567890-", "=)\:)[&# " ва шунга ўхшашлар. Баъзан ҳарфий катталиклар матнли ўзгарувчилар деб аталади ва улар орқали ЭҲМдан олинаётган маълумотлар изоҳланади. Битта матнли катталиклнинг энг катта узунлиги 255 белгидан ошмаслиги шарт. Ҳарфий катталикларнинг қийматлари " белгисидан ташқари ҳамма белгилар бўлиши мумкин.

Агар ҳарфий катталиқ " белгиларини ўз ичига олса, у ҳолда " (қўштирноқ) ичида жойлашган қўштирноқни (апостроф) белгиси билан алмаштириш керак. Масалан, агар ҳарфий катталиклнинг қиймати "ПРАВЕЦ" га тенг деб олинса, у ҳолда компьютерга " 'ПРАВЕЦ' " деб киритилади.

Ҳарфий катталиклар номи алоҳида белги билан, "Правец-8" компьютерлари учун \$ белгиси, "Агат" компьютерлари учун эса α белгиси билан яқунланади. Масалан: В5\$, М\$, PROGR\$ - "Правец-8" учун, "Агат" учун бу ҳарфий катталиклар қуйидагича ёзилади - В5 α , М α , PROG α . Қуйида барча ҳолларда ҳарфий катталиклар \$ белгиси билан келтирилади, АГАТ компьютеридан фойдаланувчи китобхонлардан мазкур компьютерларда \$ белгиси ўрнида α белгиси ишлатилишини ёдда тутишларини сўраймиз.

Ҳақиқий ва бутун ўзгарувчилар мос равишда фақат ҳақиқий ва бутун ўзгармасларни қабул қилиши керак. Ҳарфий ўзгарувчилар эса фақат ҳарфий ўзгармасларни қабул қилиши мумкин.

1.3. Стандарт математик функциялар.

Жуда кўп ҳолларда масала ва мисолларни ечиш учун дастур тўзаётганда бир хил турдаги функциялар тез-тез учраб туради. Масалан, тригонометрик, логарифмик, экспонента, илдиз

чиқариш ва шунга ўхшаш функциялар. Ҳар бир ЭҲМ учун бундай функцияларни ҳисоблаш дастурлари тузилган ва ЭҲМ хотирасига киритиб қўйилган, улар стандарт функциялар деб аталади. Бу функцияларни Бейсик тилида белгилаш учун махсус сўзлар ишлатилади. Функцияларнинг аргументлари функциянинг номидан кейин албатта қавс ичига олиб ёзилиши шарт.

Тригонометрик функцияларнинг аргументлари радиан ўлчов бирлигида берилиши керак. Агар бурчак градусларда берилган бўлса, у ҳолда уни радианга ўтказиш учун қуйидаги формула ишлатилади:

$$\text{радиан} = \text{градус} \cdot \pi / 180$$

Баъзи математик стандарт функцияларнинг Бейсик тилида ифодаланиши қуйидаги жадвалда келтирилган:

Функция	Бейсикда ёзилиши	Изоҳ
x	ABS()	Аргументнинг абсолют қиймати
arctg x	ATN(X)	Натижа радианда
cos x	COS(X)	Аргумент радианда
sin x	SIN(X)	Аргумент радианда
tg x	TAN(X)	Аргумент радианда
\sqrt{x}	SQR(X)	$x \geq 0$
e^x	EXP(X)	$x \leq 88$
ln x	LOG(X)	$x > 0$
sign x	SGN(X)	Аргументнинг ишорасини аниқлаш
[x]	INT(X)	Соннинг бутун қисми

Натурал логарифмдан ихтиёрий а асосли логарифмга ўтиш учун $\log_a b = \ln b / \ln a$ формуласидан фойдаланилади.

Бейсикда бу функциялардан ташқари, тасодифий сонларни чиқарадиган махсус функция мавжуд. Бу функция RND дир. RND функциясининг аргументи сифатида 0 сони ёки ҳақиқий сон (мусбат, манфий) бўлиши мумкин. Бу функция қуйидагича ёзилади:

RND (аргумент)

Қуйидаги мисолларни бажариб кўринг. Дастурни бажарилиш жараёнида тўхтатиб, қайтадан ишга туширинг, яъни RUN буйруғини киритинг. Дастур ишини тўхтатиш учун CTRL ва RESET тугмаларини босинг.

1-мисол: Қуйидаги дастур сатрларини клавиатура ёрдамида киритиб, 30-сатрдан кейин RUN буйруғини ёзиб, (Return) тугмасини босинг.

```
10 Z=RND(0)
20 PRINT Z
30 GOTO 10
```

2-мисол: Қуйидаги дастурни 1-мисолга ўхшаб бажаринг. Экранга чиқарилаётган сонларни таҳлил қилинг.

```
10 Z=RND(5)
20 PRINT Z
30 GOTO 10
```

1.4. Арифметик ифодалар.

Арифметик ифода деганда доимий сонлар, ўзгарувчилар, стандарт функцияларнинг арифметик амал ишоралари ("+"-қўшиш, "-"-айириш, "*" -кўпайтириш, "/"-бўлиш, "^"-даражага кўтариш белгилари) ва қавсларнинг маълум қоида асосида ёзилган кўриниши тушунилади. Арифметик ифодалар бир сатрга ҳеч қандай қўшимча сатр усти ва ости белгиларисиз ёзилади. Арифметик ифодаларнинг ёзилишига мисоллар кўрамиз:

Математикада	Бейсикда
$(a-b)(a+b)$	$(A-B)*(A+B)$
$\arctg\left(\frac{x}{2}\right)-2$	$ATN(X/2)-2$
$\sqrt{\sin 2x \cos 3x}$	$SQR(SIN(2*X)*COS(3*X))$
$e^{x+ x }$	$EXP(X+ABS(X))$

Юқоридаги ёзувлардан кўриняптики, Бейсикдаги ёзувлар математик ёзувларга жуда яқиндир, лекин математикадаги "*" -кўпайтириш ўрнига Бейсикда "*" белгиси, ":" бўлиш ўрнига Бейсикда "/" белгиси ишлатилади. Даражага кўтариш амали Бейсикда "^" белгиси орқали ифодаланади. Бейсик дастурлаш тилида дастур тузаётганда, формулалардан фойдаланишда асосан қуйидаги қоидаларга риоя қилиш керак:

1. Арифметик ифодаларни бажараётганда уларнинг бажарилиш тартиби:

- қавс ичига олинган ифодалар;
- стандарт функциялар;

- даражага кўтариш амали;
- кўпайтириш ва бўлиш амаллари;
- қўшиш ва айириш амаллари.

2. Бажарилиш тартибида бир ўринда турган амаллар (масалан, кўпайтириш ва бўлиш ёки қўшиш ва айириш амаллари) уларнинг ифодада жойлашиш тартибида чапдан ўнгга қараб бажарилади.

3. Амаллар белгиларини тушириб ёзиш (масалан, "*" белгисини), уларнинг бир нечтасини бирин-кетин ёзиш мумкин эмас.

4. Очилган қавслар сони ёпилган қавслар сонига тенг бўлиши керак.

Қуйидаги $\sqrt{ab/b+d^g+n}$ арифметик ифоданинг бажарилиш тартибини кўрамыз:

1. G+N
2. D^(G+N)
3. A*N
4. A*N/B
5. A*N/B+D^(G+N)
6. SQR(A*N/B+D^(G+N))

2. Бевосита ҳисоблаш тартиби

ЭҲМнинг бевосита ҳисоблаш тартибида ишлаш деганда ЭҲМда дастур тузмасдан сонли ахборотлар билан ишлашни тушуниш керак. ШЭҲМларнинг бевосита ҳисоблаш тартибида ишлашини кўриб чиқамиз. Бунда компьютер микрокалькулятор сифатида, яъни сонлар билан арифметик амаллар бажариш, оддий функцияларни ҳисоблаш, баъзи формулаларни ишлатиб мисоллар ечиш учун ишлатилади.

"Правец-8" компютери ишга туширилгач, экранда] белгиси, "Агат" компютери ишга туширилгач, }_ белгиси чиқади. Бу эса компютер буйруқ кутиш ҳолатида эканлигини билдиради, яъни компютер клавиатура орқали киритиладиган буйруқларни бажаришга тайёр эканлигини билдиради. Бизга баъзи бир арифметик ифодаларни ҳисоблаш керак бўлсин. Бунинг учун қуйидаги босқичларни бажариш керак бўлади:

1. ? белгиси босилади;
2. Арифметик ифода терилади;
3. Return тугмаси ("Агат" компютерерида ←) босилади.

Шуни таъкидлаб ўтиш зарурки, арифметик ифодалардан кейин = (тенглик) белгиси қўйилмайди. Яъни умумий кўринишда:

? арифметик ифода

Return тугмаси босилгандан кейин арифметик ифоданинг қиймати кейинги сатрда чиқади.

Арифметик ифодаларни ҳисоблаётганда ШЭҲМ ларнинг баъзи имкониятларини ҳисобга олиш керак. Барча дастурлаш тилларида каср сонларни ёзишда бутун сонни каср қисмидан ажратиб турадиган вергул ўрнига нуқта белгиси ишлатилади. Сонларнинг разряд хоналари 13 та хонагача автоматик ўрнатилади (берилади). Агар Сиз ишлатадиган сонлар шу разрядларда бўлса унда яхши, акс ҳолда сонларни асоси 10 кўрсаткичли (экспонент) кўринишда бериш ўринлидир.

Масалан, -0,00000004321; 123456789 сонларини қуйидагича ёзиш мумкин: -0.4321E-7; 1.23456789E+8

Қуйида фойдаланувчининг ЭҲМ билан мулоқоти кўринишда бир неча мисолларни кўриб чиқамиз.

1-мисол. 24 ва 12 сонларининг йиғиндисини топинг.

Фойдаланувчи: ? 24+12

ЭҲМ: 36

2-мисол. $A=22.5$, $B=6.78$ бўлганда $A^2 - B^2$ ифоданинг қиймати ҳисоблансин.

Фойдаланувчи: ? $22.5^2 - 6.78^2$

ЭҲМ: 460.2816

Бевосита тартибда ишлаётганда стандарт математик функциялардан ҳам фойдаланиш мумкин.

3-мисол. $\pi=3.14159$ бўлганда $e^{\sin\pi/4}$ ни ҳисобланг.

Фойдаланувчи: ? EXP(SIN(3.14159/4))

ЭҲМ: 2.0281140302731

Бевосита ҳисоблаш тартибида ишлаётган вақтда хатоларга йўл қўйилиши табиий. Мабодо арифметик ифодаларни ёзаётганингизда хатоликка йўл қўйганингизни сезиб қолсангиз

ва Return тугмаси босилмаган бўлса, у ҳолда Сиз шу хатони тузатишингиз мумкин. Бунинг учун «←, →» тугмалардан фойдаланиб хатони тузатиш керак ва сатр охиригача "ўтиб", сўнгра Return тугмасини босиш керак.

4-мисол. Қуйидаги ифода ҳисоблансин:

$$\sqrt{3.3\sin\pi/2+1}$$

Фойдаланувчи: ? SQR(3.3SIN(3.14159/2)+1)

ЭҲМ: ? Syntax error (Агат компютерида: Синт. ош. в) — Синтактик хато.

Бу хабар фойдаланиш мумкин бўлмаган белги киритилганидан далолат беради. Арифметик ифодаларни ёзаётган пайтда ҳеч бир белгини ёзмалик (ташлаб кетиш) мумкин эмас.

Юқоридаги мисолда 3.3 ва SIN(3.14159/2) орасида кўпайтириш белгиси қўйилмаган, бу эса хатоликка олиб келган. Хатони тузатиб, Return тугмасини босак, компютер экранида қуйидаги натижани ҳосил қиламиз:

2.0736644

5-мисол. $X=3$ да қуйидаги ифоданинг қийматини ҳисобланг:

$$(6x^2-3x+5)/(x^2-4x+3)$$

Фойдаланувчи: ? (6*3^2-3*3+5)/(3*3-4*3+3)

ЭҲМ: Division by zero error (Агат компютерида: X/0) — Нольга бўлиш, яъни, касрнинг махражида 0 сони ҳосил бўлиб, 0 га бўлишга ҳаракат қилиняпти.

Тригонометрик функцияларнинг қийматларини ҳисоблаётганда уларнинг қийматлари радианда берилишини айтиб ўтиш зарур. Градусдан радианга ўтиш учун

$$\text{радиан} = \text{градус} * 3.14159/180$$

формуласидан фойдаланиш керак.

6-мисол. Sin 30° ҳисоблансин.

Бунинг учун аввал 30 градусни радианга юқоридаги формула бўйича ўтказамиз:

Фойдаланувчи: ? 30*3.14159/180

ЭҲМ: 0.5235983

Шундан сўнг $\text{Sin } 0,5235983$ ни ҳисоблаймиз:

Фойдаланувчи: ? $\text{SIN}(0.5235983)$

ЭҲМ: 0.4999996

Шундай бўлиши мумкинки, арифметик ифоданинг қийматини ҳисоблаётган вақтда белгилар сони 255 тадан ошиб кетиши мумкин. Бундай ҳолда арифметик ифодани кичикроқ ифодаларга бўлиб, ҳар бирини алоҳида-алоҳида ҳисоблаш тавсия этилади.

7-мисол. $\sqrt{\text{SIN}^2/3-1}$ нинг қиймати ҳисоблансин.

Фойдаланувчи: ? $\text{SQR}(\text{SIN}(3.14159/3)-1)$

ЭҲМ: Illegal quantity error (Нотўғри функционал чақириш)

Фойдаланувчи: ? $\text{SIN}(3.14159/3)$

ЭҲМ: 0.8660249

Ҳақиқатан ҳам, $0,866025 - 1 = -0,133975$ - манфий сон бўлганлиги учун машина илдиз чиқара олмайди. Бевосита тартибда ҳисоблашлардан кейин ифоданинг матни машина хотирасида сақланмайди.

3.Бейсик тилининг асосий буйруқлари

Фойдаланувчи ЭҲМга барча фармойишларини Бейсик тилининг маълум бир ибораларидан тузилган кўринишда беради. Бу иборалар икки хил: буйруқ ва операторлардан иборат. Буйруқ ва операторлар ўртасида катта фарқ йўқ: буйруқлар кўпинча бевосита тартибда, операторлар эса дастур таркибида ишлатилади. Лекин барча буйруқларни ҳам дастур таркибида, операторларни эса (бир неча оператор бундан мустасно) бевосита тартибда қўллаш мумкин. Буйруқлар кўп ҳолларда бевосита тартибда клавиатура ёрдамида сатр номерисиз ёзилиб, RETURN тугмаси босилиши билан бажарилади. Қуйида асосий буйруқларни кўриб чиқамиз:

NEW - бу буйруқ компьютернинг ўзгарувчан хотирасини тозалайди, яъни хотирада мавжуд бўлган дастурни ўчиради.

Бу буйруқни бажариш учун клавиатурадан NEW ибораси киритилади ва RETURN тугмаси босилади.

HOME буйруғи дисплей экранини тозалайди ва курсор экраннинг чап юқори бурчагига бориб туради. Буни бажариш учун HOME ибораси киритилади ва RETURN тугмаси босилади.

LIST буйруғи дисплей экранига ўзгарувчан хотирадаги дастур матнини энг кичик номеридан бошлаб чиқариш учун хизмат қилади. Дастур матни экранга чиқарилаётганда ўқиш учун қулай кўринишда чиқарилади. Шунинг учун киритилган дастур билан чиқарилаётган дастурда салгина фарқ бўлиши мумкин. Масалан,

```
10 ?"САЛОМ"
```

сатрини киритгандан кейин, LIST оператори ёрдамида дастур қайта чақирилса, экранга

```
10 PRINT "САЛОМ"
```

кўринишида чиқарилади. Агар дастурнинг матни жуда узун бўлса, у ҳолда LIST буйруғининг турли кўринишларидан фойдаланиш керак. Агар Сиз фақат битта N-сатрни кўрмоқчи бўлсангиз, у ҳолда қуйидагича

```
LIST N
```

кўринишида буйруқ берасиз. Бу буйруқнинг бажарилишини кўриш учун қуйидаги дастурни ёзамиз (ихтиёрий сон кубини ҳисоблаш):

```
5 REM Сонларни кубга кўтариш
10 INPUT"СОН";A
20 ?"КУБ=";A^3
30 INPUT"Давом этасизми ? (х/й)";A$
40 IF A$="х" THEN 10
50 END
```

Фараз қиламизки, 20-сатрни ўзгартириш керак. Фақат 20-сатрни чақириниш учун LIST 20 буйруғини берамиз. Шунда экранга

```
20 PRINT"КУБ=";A^3
```

сатри чиқарилади.

Агар экранда бир нечта сатрни кўрмоқчи бўлсангиз, яъни N-сатрдан M-сатргача, унда

```
LIST N, M
```

буйруғини беришингиз керак.

Агар юқоридаги мисол учун LIST 20, 40 буйруғини берсак, унда экранда

```
20 PRINT"КУБ=";A^3
30 INPUT"Давом этасизми?(х/й)";A$
40 IF A$="х" THEN 10
```

сатрлар берилади.

Дисплей экранда дастурни N-чи сатрдан бошлаб охиригача кўрмоқчи бўлсангиз, унда

```
LIST N,
```

кўринишда буйруқ бериш керак. Юқоридаги дастур учун LIST20, буйруғини берсак, унда 20-сатрдан бошлаб барча сатрлар кўрсатилади.

```
20 PRINT"КУБ=";A^3
30 INPUT"Давом этасизми?(х/й)";A$
40 IF A$="х" THEN 10
50 END
```

Агар Сизга бир бошдан N-номергача барча сатрларни кўриш керак бўлса, унда LIST,N буйруғини бериш керак. Юқоридаги мисолда 20-сатргача бўлган сатрларни кўриш учун LIST,20 буйруғини берсак, унда дисплей экранига

```
5 REM Сонларни кубга кўтариш
10 INPUT"СОН";A
20 PRINT"КУБ=";A^3
```

сатрлари чиқарилади.

DEL N, M буйруғи ўзгарувчан хотирадаги дастурдан N-сатрдан бошлаб M-сатргача бўлган қисмини олиб ташлайди (ўчиради). Масалан, қуйидаги

```
10 INPUT A
20 INPUT B
30 PRINT A*B
```

```
40 PRINT A+B  
50 PRINT A/B  
60 END
```

дастурни хотирага киритиб, DEL 40, 60 буйруғини бажарсак ва LIST буйруғини бериб, экранга қарасак, унда 40, 50 ва 60 сатрлар йўқолиб экранда

```
10 INPUT A  
20 INPUT B  
30 PRINT A*B
```

сатрлар қолганини кўрамиз.

Агар бир сатрни йўқотмоқчи бўлсак, у ҳолда DEL N, N буйруғини бериш кифоя. Фақат бир сатрни бошқача, яъни шу сатр номерини ҳеч қандай матн киритмай териш ва Return тугмасини босиб ҳам йўқотиш мумкин.

RUN буйруғи ўзгарувчан хотирада мавжуд бўлган дастурни бажариш учун хизмат қилади. RUN буйруғидан кейин компьютер дастурни энг кичик номерли сатрдан бошлаб бажаришга киришади. Бу буйруқнинг умумий кўриниши қуйидагича:

```
RUN
```

Агар берилган дастурни N-сатрдан бошлаб бажариш зарур бўлиб қолса, у ҳолда RUN буйруғининг

```
RUN N
```

кўринишидан фойдаланилади. Бу ерда N – бажариш керак бўлган сатр номери. Масалан:

```
RUN 30
```

Бу ҳолда дастур 30-сатрдан бошлаб бажарилади.

4.Бейсик тилининг асосий операторлари

Компьютер фойдаланувчи томонидан қўйилган масалани муфассал, аниқ ва тушунарли кўрсатмалар берилгандагина бажара олади. Бу кўрсатмалар маълум бир маънони англатувчи сўзлардан иборат бўлиб, компьютерга қандай операцияни бажариш лозимлигини билдиради ва операторлар деб аталади. Бундай операторлар кетма-кетлиги дастурни ҳосил қилади. Дастурда операторлар сатрларга бўлиб ёзилади, бир сатрга бир ёки бир нечта оператор ёзиш мумкин.

Бейсик дастурлаш тилида (қисқача, Бейсик тилида) ҳар бир сатр номердан бошланиши керак. Бу сатр номери чегаралари ҳар бир компьютер учун ҳар хил бўлиб, "Правец-8" ва "Агат" ЭХМлари учун

$$0 \leq \text{сатр номери} \leq 63999.$$

Компьютер ҳар бир дастурни сатр номерининг ўсиб бориш тартибида бажаради. Сатр номери бошқарувни бир сатрдан иккинчи сатрга узатиш учун ҳам керак. Сатрлар асосан 10 қадам билан номерланадилар (10, 20, 30, ...). Бу усул дастурлашда катта аҳамиятга эга, чунки керак бўлиб қолган тақдирда икки сатр орасига яна қўшимча сатрларни қўйиш мумкин. Ҳар бир сатр Return тугмасини босиш билан тугалланади. Бу тугма босилгандан кейин ушбу сатр компьютер хотирасига киритилади.

Сатрларнинг максимал узунлиги 255 белги. Экраннынг ҳар бир сатрига 40 та белги сиғади. Агар киритилаётган белгилар сони 40 та белгидан ошса, бу белгиларни компьютер автоматик равишда иккинчи сатрга ўтказади.

Дастур сатр номерининг ўсиб бориш тартибида бажарилади. Баъзида бажарилаётган сатрда бошқарув бошқа сатрга ўтказилган бўлиши ҳам мумкин (дастурлаш қоидаларига кўра, GOTO, ON GOTO, GOSUB, IF операторларига қаранг).

Ҳар бир сатрда бир ёки бир нечта оператор ёзилади. Агар бир нечта оператор ёзилса, у ҳолда улар бир биридан икки нуқта билан ажратилиб ёзилади. Масалан:

```
10 A=12.5
20 B=56
30 C=45
```


40 D=A-B+C
50 PRINT D
60 END

Бу дастур қуйидаги дастурга тенг кучли:

10 A=12.5 : B=56 : C=45 : D=A-B+C
20 PRINT D
30 END

Бу дастурнинг 10-сатрида A, B ва C ўзгарувчиларига уларнинг қийматлари мос келтириляпти ва шу сатрнинг ўзида A-B+C нинг қиймати ҳисобланиб, у D ўзгарувчининг қийматига тенглаштириляпти. 20-сатрда D ўзгарувчининг қиймати дисплей экранига чиқариляпти. 30-сатр дастур якунини кўрсатади.

4.1. Изоҳ оператори - REM.

Айрим пайтда дастурларни ҳужжатлаштиришга, баъзи изоҳларни киритишга тўғри келади. Бундай операцияни амалга ошириш учун REMARK (изоҳ) ёки қисқача REM оператори ишлатилади. Бу операторнинг умумий кўриниши

m REM матн

бўлиб, бу ерда:

m - сатр номери,

REM - операторнинг номи,

"матн" - ихтиёрий сўз ёки гап бўлиши мумкин.


REM операторидан кейин келадиган матнда сатр охиригача ихтиёрий сўз ёзиш мумкин. Бу операторни компьютер бажармайди.

Шуни ҳам айтиш муҳимки, бу оператордан кейин бир сатрда операторларни ишлатиш мумкин эмас. REM операторини бошқа операторлар билан бир сатрда ишлатиш мумкин, лекин REM сатрда энг охириги оператор бўлиши керак. Масалан:

10 REM y=Sinх функцияни ҳисоблаш.

15 X = 0 : REM X ўзгарувчига 0 қиймат бериляпти.

Дастурда ишлатилаётган REM оператори қанча кўп бўлса, у шунчалик тушунарли бўлади.



4.2. LET-ўзлаштириш оператори.

Бу операторнинг умумий кўриниши қуйидагича:

m LET Y=A

Бу ерда

m - сатр номери,

LET - операторнинг номи бўлиб, "бўлсин" маъносини билдиради,

Y - ўзгарувчининг номи бўлиб, унга A ифоданинг қиймати берилади.

Ўзгарувчининг номи сифатида ихтиёрий янги ёки илгари ишлатилган ўзгарувчи, ифода сифатида - ихтиёрий ифодани ишлатиш мумкин.

Операторнинг бажарилиши: Бу оператордаги A ифоданинг қиймати топилади ва унинг қиймати Y ўзгарувчининг қиймати сифатида қабул қилинади.

1-мисол. A, B, C, D ўзгарувчилар киритилсин. A=10, B=2, C=5, D=3 деб олиниб, уларнинг йиғиндиси топилсин.

```
10 LET A=10
20 LET B=2
30 LET C=5
40 LET D=3
50 LET Y=A+B+C+D
```

ёки бу сатрларни битта сатрга ҳам ёзиш мумкин:

```
10 LET A=10:LET B=2:LET C=5:LET D=3:LET Y= A+B+C+D
```

2-мисол. A=8.3 ва B=9.8. X=A+B+A/B нинг қиймати ҳисоблансин.

```
10 LET A=8.3
20 LET B=9.8
30 LET X=A+B+A/B
```

Бу операторларнинг бажарилиши натижасида X ўзгарувчи 18.9469387755 қийматни қабул қилади.

LET оператори дастурнинг ихтиёрий қисмида, ихтиёрий операторлар билан биргаликда ёзилиши мумкин. Бу компьютерларда LET ёрдамчи сўзини ёзмаслик ҳам мумкин. У ҳолда LET операторининг умумий кўриниши қуйидагича бўлади:

m Y=A

Ўзлаштириш операторларининг ёзилишига мисоллар:

```
10 S=S+X*Y
20 A=B^2 - C^2
30 A$="МАТЕ"+"МАТИКА"
```

Ҳақиқий ўзгарувчига ҳарфий қиймат ёки ҳарфий ўзгарувчига ҳақиқий қиймат мос келтирилса, у ҳолда система хатолик ҳақида маълумот беради: TYPE MISMATCH ERROR IN mm (mm сатрда типлар мос келмайди). Масалан:

```
10 A=223 : B=A - 56
20 C = "ИНФОРМАТИКА"
30 Y=A +B+C
RUN
TYPE MISMATCH ERROR IN 20 (Агат компьютерида: Ош. тип в 20...)
```

Бу дастурнинг 20-сатрида ҳақиқий ўзгарувчига ҳарфий катталиқ мос келтириляпти. Бу эса мумкин эмас, шунинг учун система 20-сатрда хатоликка йўл қўйилганини хабар қиляпти.

4.3. PRINT оператори.

PRINT оператори орқали компьютердаги турли ахборот: графиклар, ўзгарувчиларнинг қийматлари, натижалар ва матнлар дисплей экранига (ёки принтерга) чиқарилади. PRINT операторининг умумий кўриниши қуйидагича

m PRINT a1, a2, ...

бўлиб, бу ерда

m – сатр номери;

a1, a2, ... – қиймати чиқарилиши керак бўлган объектлар рўйхати. a1, a2, объектлар бир-биридан ",", "ёки ";" билан ажратилади. Масалан,

```
PRINT A
PRINT A^2 -B^2
PRINT"Жуда яхши"
PRINT W$
```

Агар PRINT оператори орқали қиймати чиқарилаётган объект сифатида арифметик ифода бўлса, у ҳолда бу арифметик

ифоданинг қиймати ҳисобланади ва натижаси рақамлар ёрдамида чиқарилади. Мусбат сонларни чиқаришда "+" белгиси сон олдида чиқарилмайди. Бундан ташқари "ортиқча" ноллар ҳам чиқарилмайди. Масалан қуйидаги қаторни экранга ёзиб, уни бажарсак

```
PRINT 1.2998 * 0.0000125  
1.52475E - 05
```

ёзувни кўрамыз.

Ҳарфий катталиклар чиқарилаётганда ташқи қўштирноқлар чиқарилмайди. Масалан,

```
PRINT "Правец-8А"  
Правец-8А
```

Агар чиқарилаётган ахборот кўп бўлса, улар вергул ёки нуқтали вергул билан бир-биридан ажратилади. Агар ифодаларларни ажаратишда ";" ишлатилган бўлса, у ҳолда PRINT операторидаги биринчи ифода қиймати экран сатрининг чап томонида босиб чиқарилади, ундан сўнг бўш жой ташланмасдан кейинги ифода қиймати босиб чиқарилади.

1-мисол. Қуйидаги соннинг даражаларини чиқарадиган дастурни компьютерга киритиб, бажариб кўрамыз:

```
10 LET A=-13  
20 PRINT A;A^2;A^3  
RUN  
-13169-2197
```

Юқорида айтганимиздек, экрандаги бу -13169-2197 сон ягоналиги ёки A, A², A³ ларнинг натижасими, билиш жуда қийин.

2-мисол. Берилган учта 100; 8; 8,5 сонларини дисплей экранига чиқаринг.

```
10 A=100 : B=8 : C=8.5  
20 PRINT A ; B ; C
```

Дастур бажарилгандан кейин берилганлар қуйидаги кўринишда чиқарилади:

```
ўринлар 123456789 . . .  
10088.5
```

Бу мисолда ҳамма сонлар мусбат бўлгани учун уларни босиб чиқаришда соннинг ишораси "+" чиқарилмаган.

3-мисол. Берилган "ИН", "ФОР", "МА", "ТИ", "КА" ҳарфий ўзгармасларни экранга чиқаринг.

```
10 PRINT "ИН";"ФОР";"МА";"ТИ";"КА"
RUN
```

Дастур бажарилиши натижасида экранда ИНФОРМАТИКА ёзувини ҳосил қиламиз.

Берилганларни чиқаришда дисплей экрани бир неча соҳага ажралади. Бу соҳаларнинг узунлиги 16 та ўриндан иборат бўлади. Бошқача айтганда 1 та соҳага 16 тагача белги ёзилиши мумкин, яъни

- 1-соҳа 0-16 ўринлар
- 2-соҳа 17-32 ўринлар
- 3-соҳа 33-40 ўринлар.

Берилганларни босиб чиқаришда қуйидагича қоида мавжуд. Чиқарилиши керак бўлган ифодалар бир-биридан ", " билан ажратилган бўлса, ", " белгиси чиқарилаётган ифодани кейинги соҳага чиқарилиши ҳақида кўрсатма беради. Агар ", " билан ажратилган объектлар 3 тадан ортиқ бўлса ва 3-ифоданинг қийматидаги белгилар сони 8 тадан ортиқ бўлса, у ҳолда 3-ифоданинг қолган қийматлари ҳеч қандай жой ташланмасдан 1-ифоданинг пастки сатрида босиб чиқарилади.

4-мисол. Қуйидаги дастур сатрларини бажариб кўринг:

а) 10 PRINT "Баҳоингиз",5

Дастур бажарилиши натижасида PRINT оператори аргумент-лари экранда қуйидагича жойлашади:

ўринлар	1	17
	Баҳоингиз	5

б) 20 PRINT "Берилган", "дастурнинг", "бажарилишига", "мисол"

Бу оператор бажарилганда

ўринлар	1	17
	Берилган	дастурнинг
	бажарилишига	мисол

ёзувлари босиб чиқарилади.

Ҳарфий катталиқлар ҳар қандай шaroитда ҳар бир соҳанинг 1-ўрнидан бошлаб босиб чиқарилади.

5-мисол. Қуйидаги сатрларни бажариб кўринг.

```
10 A=5 : B=6 : C$="A"  
20 PRINT A, B, C$
```

Дастурнинг бажарилиши натижасида А, В ва С\$ ларнинг қийматлари қуйидаги ўринларда чиқарилади:

ўринлар	1	17	33
	5	6	A

Бейсикда PRINT ёзуви ўрнига "?" сўроқ белгиси ишлатилиши мумкин. Масалан:

```
10 ? "САЛОМ"
```

Агар PRINT оператори орқали илгари аниқланмаган сонли ўзгарувчи босиб чиқарилса, у ҳолда бу сонли ўзгарувчига 0 қиймат берилади. Масалан :

```
10 A=10: B=-5  
20 ? A, B, C
```

дастурни бажарсак, дисплей экранида C нинг қиймати 0 га тенг бўлганини кўрамиз:

ўринлар	1	17	33
	10	-5	0

Босиб чиқарилаётган пайтда ортиқча битта соҳа ташлаб юбориш учун, битта ", " белгиси қўйилиши керак. Масалан:

```
10 A=-7: B = 6: C$="A"  
20 ? A, B, , C$
```

Натижа:

ўринлар	1	17	33
	-7	6	
	A		

6-мисол. $x=10$ бўлганда $y=\sin x$ функциянинг қиймати ҳисоблансин.

```
10 X=10
20 Y=SIN(X)
30 PRINT "X="; X; " да",
40 PRINT "SIN(X)="; Y
```

Дастур бажарилгандан кейин дисплей экранида натижа босиб чиқарилади:

```
X=10 да SIN(X)= .54402111
```

PRINT оператори орқали қиймати босиб чиқарилаётган ҳарфий ўзгарувчидан кейин сонли ўзгарувчи келса, ";" белгиси ёзилмаслиги мумкин. Бундай ҳолда ҳам сонли ўзгарувчининг қиймати ҳарфий ўзгарувчининг қийматидан кейин бўш жой ташланмасдан босиб чиқарилади. Масалан, юқоридаги 6-мисолда 30- ва 40-сатрларни қуйидагича ўзгартирсак

```
30 PRINT "X="X;" ДА",
40 PRINT "SIN(X)="Y
```

ва қайтадан бажарсак, у ҳолда дисплей экранида қуйидаги натижани ҳосил қиламиз:

```
X=10 даSIN(X)= .54402111
```

Босиб чиқарилаётган қийматлар орасида бир ёки бир неча бўш сатр ташлаш мумкин. Бунинг учун дастурга қўшимча PRINT оператори киритилади.

4.4. INPUT оператори.

Бу оператор дастур бажарилиш жараёнида компьютерга берилганларни киритиш учун хизмат қилади. Операторнинг умумий кўриниши:

```
m INPUT a1, a2, . . .
```

бу ерда:

m — сатр номери;

a1, a2, . . . — жадвал элементлари, ўзгарувчилар номи

(ўзгарувчилар рўйхати), улар бир-биридан ", " билан ажратилади. Масалан,

```
100 INPUT A
105 PRINT A
110 INPUT K$
120 PRINT K$
```

INPUT оператори бажарилганда дастур бажарилиши вақтинча тўxtатилади, компьютер экранида "?" белгиси ва ёниб-ўчиб турган курсор пайдо бўлади, система фойдаланувчининг ўзгарувчига конкрет қиймат беришини кутиб туради. Фойдаланувчи қийматларни киритиб, RETURN тугмасини босгач, қийматлар дисплей экранида "?" белгисидан кейин намоён бўлади ва дастур бажарилиши давом эттирилади.

Киритилаётган қиймат INPUT операторида ёзилган ўзгарувчининг турига мос бўлиши керак. INPUT операторидан кейин сонли ўзгарувчи бўлса, киритилаётган қиймат сонли бўлиши, агар матнли ўзгарувчи бўлса, киритилаётган қиймат матнли бўлиши керак. Агар киритилаётган қиймат тури ўзгарувчининг туридан фарқли бўлса, у ҳолда дисплей экранида

? REENTER (бошқатдан киритинг) [Агат компьютерларида Ош. ввода-киритиши хатоси]

хабари чиқарилади ва қайтадан "?" белгиси пайдо бўлади.

Ўзгарувчининг қийматида қўштирноқ ва вергул белгилари ишлатилмаса, у ҳолда унинг қийматини қўштирноқ ичига олиш шарт эмас.

1-мисол. INPUT операторидан фойдаланиб, Юлдуз сўзини экранга чиқаринг.

```
10 INPUT A$
20 PRINT A$
RUN
? Юлдуз
Юлдуз
```

Шуни айтиб ўтиш керакки, матнли ўзгарувчиларнинг қийматларини киритишда " " белгиси ишлатилмаса, у ҳолда қийматнинг чап томонида киритилаётган ҳамма пробел (бўш жой)лар ташлаб юборилади. Юқоридаги дастурни яна бир марта қуйидагича бажарамиз:


```

RUN
? Юлдуз
Юлдуз

```

Агар бу мисолда " белгиси ишлатилса, у ҳолда қийматнинг чап томонида ёзилган бўш жойлар сақланиб қолади:

```

RUN
?" Юлдуз"
Юлдуз

```

Битта INPUT операторида бир-биридан вергул белгиси билан ажратилган ихтиёрий типли бир неча ўзгарувчиларнинг қийматларини киритиш мумкин. Масалан,

```
INPUT A, KO, F$, E.
```

Бу операторнинг бажарилишида фойдаланувчи киритилаётган ўзгарувчиларнинг қийматларини ", " билан ажратиб, битта сатрда ёзиши мумкин:

```
? 1, 32, СОН, 5
```

Агар матнли ўзгарувчининг қийматида ", " белгиси бўлса, компьютер уни ўзгарувчиларни бир-биридан ажратувчи белги сифатида олмаслиги учун бу ўзгарувчининг қийматини " белгиси ичига олиш керак. Масалан, ? 1991, 32, "СОН, БУТУН", 5.

INPUT оператори бажарилаётганда ўзгарувчилар қандай тартибда ёзилган бўлса, уларнинг қиймати ҳам шу тартибда киритилиши керак. Агар INPUT операторида ўзгарувчилар номлари кўп бўлса ва фойдаланувчи биттасининг қийматини киритиб, Return тугмасини босса, у ҳолда пастки сатрда "??" белгиси пайдо бўлади ва компьютер қолган ўзгарувчиларнинг қийматлари киритилишини кутиб туради. Фойдаланувчи қолган ўзгарувчиларнинг қийматларини киритиши мумкин.

2-мисол. А\$, В, С\$, D, Е\$ турли ўзгарувчиларни ва В ва D кўпайтмасини экранга чиқаринг.

```

10 INPUT A$, B, C$, D, E$
20 PRINT A$:B
30 PRINT C$:D
40 PRINT E$;" ";B*D

```

10-сатрда 5та ўзгарувчи А\$, В, С\$, D ва Е\$ ларнинг

қийматини клавиатура ёрдамида киритиш мўлжалланган бўлиб, булар мос равишда матнли, сонли, матнли, сонли ва матнлидир. 20-сатрда биринчи жуфт ўзгарувчиларнинг қийматлари чиқарилади, 30-сатрда 2-жуфт ўзгарувчиларнинг қийматларини чиқариш кўзда тутилган. 40-сатрда эса E\$ ўзгарувчининг қиймати билан B ва D қийматларнинг кўпайтмаси чиқарилиши дастурлаштирилган. Бу дастурни бажарамиз:

```
RUN
? УЗУНЛИГИ, 147, КЕНГЛИГИ, 89, ЮЗАСИ
УЗУНЛИГИ 147
КЕНГЛИГИ 89
ЮЗАСИ 13083
```

Агар INPUT операторида бир нечта ўзгарувчи келтирилган бўлса, оператор бажарилганда киритилаётган қийматларнинг қайси ўзгарувчига мос келишини билиш қийин. Шунинг учун Бейсикда INPUT операторида матнли ўзгармасларни (изоҳларни) бериш имконияти кўзда тутилган. Бу матнли ўзгармаслар қўштирноқ ичига олиниб, улар экранда ўзгарувчилар қийматларини киритишдан олдин босиб чиқарилади ва қайси ўзгарувчининг қиймати киритилишини кўрсатиб туради. INPUT операторида изоҳдан кейин ";" белгиси қўйилиб, ўзгарувчининг (ёки ўзгарувчиларнинг) номи берилди:

```
m INPUT "изоҳ"; a1, a2, ...
```

Оператор бажарилганда экранга изоҳ матни ва ёниб-ўчиб турган курсор чиқарилади (? белгиси чиқмайди). Масалан,

```
10 INPUT "Вақтни киритинг (соат)";T
20 INPUT "Тезликни киритинг (км/соат)";V
30 PRINT "S=";V*T
```

Дастур бажарилиши натижасида дисплей экранида

```
Вақтни киритинг (соат)
```

ёзуви чиқади ва компьютер T ўзгарувчининг қиймати киритилишини кутиб туради. T нинг қийматини 2 га тенг деб олиб, компьютерга киритамиз:

```
Вақтни киритинг (соат) 2
```

Return тугмаси босилгандан кейин дисплей экранида V

ўзгарувчининг қийматини киритиш лозимлигини уқтирувчи қуйидаги ёзув чиқади:

Тезликни киритинг (км/соат)

V нинг қийматини, масалан, 320га тенг деб олиб, компьютерга клавиатура орқали киритамиз:

Тезликни киритинг (км/соат) 320

Return тугмаси босилгандан кейин дисплей экранида S=640 натижа чиқади. INPUT операторида матнли ўзгармасларни ишлатиш дастур билан ишлашни анча енгиллаштиради.

3-мисол.

```
10 INPUT "Сон "; A
20 INPUT "Даража кўрсаткичи "; B
30 PRINT A ; " нинг "; B - ; " даражаси "; A^B
```

Бу дастурни бажарамиз:

```
RUN
Сон 8
Даража кўрсаткичи 3
8 нинг 3- даражаси 512
```

4.5. LINE INPUT оператори.

Матнли ўзгарувчиларнинг қийматларини киритиш учун LINE INPUT оператори ишлатилади. Бу операторнинг умумий кўриниши қуйидагича

m LINE INPUT матнли ўзгарувчи

бўлиб, бу ерда

m – сатр номери;

матнли ўзгарувчи – ихтиёрий матнли ўзгарувчи.

Бу операторнинг INPUT операторидан фарқи шундаки, INPUT орқали ихтиёрий типли ўзгарувчиларнинг қийматларини киритиш мумкин бўлса, LINE INPUT орқали фақат матнли ўзгарувчиларнинг қийматларини киритиш мумкин. Масалан,

```
10 LINE INPUT A$
20 PRINT A$
```

Дастур бажарилиши натижасида дисплей экранида курсор белгиси пайдо бўлади.

Бу пайтда фойдаланувчи ихтиёрий сўзни, ёки гапни берилган ўзгарувчи (A\$) нинг қиймати сифатида киритиши мумкин. Лекин бу сўз ёки гаптаги белги (ҳарф) лар сони 255 тадан ошмаслиги керак. Сўз ёки гапнинг киритилиш жараёнини фақат Return тугмасини босиш билан тугаллаш мумкин. Сўз ёки гапта қўштирноқ, вергул, бўш жойларни ҳам ишлатиш мумкин.

Юқоридаги мисолда RUN буйруғини берсак, у ҳолда дисплей экранида курсор пайдо бўлади ва система фойдаланувчидан A\$ ўзгарувчининг қийматини киритишни кутиб туради. Қуйидаги матнли қийматни киритамиз:

```
Erning eng yuqori chuqqisi -"ZBERECT"
```

```
Return tugmasini bossak, ekranda
```

```
Erning eng yuqori chuqqisi -"ZBERECT"
```

ёзуви пайдо бўлади.

4.6. STOP ва END операторлари.

Бу операторларнинг умумий кўриниши қуйидагича:

```
m STOP
```

```
m END
```

Бу ерда m – сатр номери.

STOP оператори ёрдамида бажарилаётган дастурни тўхтатиш мумкин. Бу ҳолда система бевосита ҳисоблаш тартибига ўтади. Масалан,

```
10 A=15 : B=1.3 : C=-6
```

```
.....  
200 STOP
```

```
.....  
300 PRINT A+B+C
```

Дастурнинг 200-сатри бажарилгач, экранга

```
BREAK IN 200
```

ёзуви чиқади. STOP оператори билан тўхтатилган дастурда ҳамма ўзгарувчиларнинг қийматлари ва берилганлар сақланади.

STOP оператори фойдаланувчига дастур бажарилишини тўхтатиб, баъзи ўзгарувчиларнинг қийматларини аниқлаш, уларни ўзгартириш имконини беради. Битта дастурда STOP ни бир неча марта ишлатиш мумкин.

END оператори (матн охири) дастурнинг бажарилишини тугатади ва системани бевосита ҳисоблаш тартибига ўтказлади. Одатда фақат шу оператор дастурни тугатади. Дастурда қисм дастурлар бўлмаса, END операторини ёзмаса ҳам бўлади. Бу ҳолда ҳам дастур охириги сатригача бажарилади.

1-мисол. $R=15$ бўлганда доиранинг юзи ҳисоблансин. Натижа дисплей экранига чиқарилсин.

```
10 LET R=15
20 LET P=3.1415
30 PRINT P*R^2
40 STOP
50 END
```

Дастур бажарилгандан кейин дисплей экранига натижани ҳосил қиламиз:

```
706.8375
BREAK IN 40
```

2-мисол. Қуйидаги ифоданинг қиймати топилсин:

$$S=2,3467+\sqrt{2a^2b^3+2a^2c^3+2b^3c^3-a^4-b^4}$$

```
10 INPUT A, B, C
20 P=2 *A^2 *(B^3+C^3)+2 *B^3 *C^3
30 P=P -A^4 -B^4
40 S=2.3467 + SQR(P)
50 PRINT S
60 END
```

Шуни ёдда сақлаш керакки, бу дастурнинг бажарилишида киритилаётган А, В, С нинг қийматига қараб илдириш остидаги сон манфий бўлиши мумкин. Бу ҳолда система хатолик тўғрисида ахборот беради.

4.7. Шартсиз ўтиш оператори GOTO.

Биз бу бўлимда GOTO (... га бор) буйруғи билан танишамиз. Бу оператор бошқарувни бир сатрдан ҳеч қандай шартсиз бошқа сатрга ўтказиш учун хизмат қилади. Операторнинг умумий кўриниши қуйидагича:

m GOTO n

Бу ерда

m – сатр номери;

n – бошқарув узатилиши керак бўлган сатр номери.

GOTO операторининг бажарилиши натижасида навбатдаги бажариладиган сатр шу операторда кўрсатилган сатр бўлади. Масалан, GOTO 20 оператор мавжуд бўлган сатр дастурнинг қаерида бўлишидан қатъий назар кейинги бажариладиган сатр 20-сатр бўлади.

1-мисол.

```
10 INPUT A, B, C
20 PRINT A+B-C
30 GOTO 10
```

Бу дастурнинг бажарилишида компьютер A, B, C ўзгарувчиларнинг киритилишини кутиб туради:

? 3, 6, 2

ЭХМ талабига мувофиқ юқоридаги мос сонларни киритиб, Return тугмасини босамиз. Экрanga бажарилган дастурнинг натижаси чиқади:

7

ва бошқарув 10-сатрга ўтказилади, яъни компьютер A, B, C ўзгарувчиларнинг янги қийматларини кутиб туради. Агар бошқарув узатилаётган сатр дастурда бўлмаса, у ҳолда экранда m-сатрда хатолик мавжудлиги ҳақида хабар чиқади:

```
UNDEF'D STATEMENT ERROR IN m (Агат компютерида: ? НЕТ НОМЕРА В...)
```

Иложи бўлса, дастурда камроқ GOTO операторларини ишлатиш керак, чунки GOTO оператори қанча кўп бўлса, дастурни ўқиш шунча қийинлашади.

4.8. Шартли ўтиш оператори IF - THEN.

Баъзи-бир шартларни текширишга боғлиқ бўлган масалаларни ечиш учун махсус шартли ўтиш оператори мавжуд. Бу оператор IF-THEN операторидир. IF-THEN оператори шартнинг

бажарилиши ёки бажарилмаслигига кўра бошқарувни керакли сатрга ўтказди. IF-THEN операторининг умумий кўриниши қуйидагича:

m IF шарт THEN сатр номери

ёки

m IF шарт THEN оператор ёки операторлар

Бу ерда

m - сатр номери;

IF (агар), THEN (у ҳолда) - Бейсикнинг хизматчи сўзлари.

Шарт текшириляётганда қуйидаги муносабатлардан биттасининг тўғри ёки нотўғрилиги текширилади:

< - кичик

<= - кичик ёки тенг

> - катта


>= - катта ёки тенг

= - тенг

<> - тенг эмас

Оператор ишлаштини кўриб чиқамиз. Дастурнинг шу оператор ёзилган сатри бажарилганда аввал шартли ифоданинг қиймати ҳисобланади. Агар шарт бажарилса, у ҳолда бошқарув THEN жумласига берилади, акс ҳолда, яъни шарт бажарилмаса, бошқарув кейинги сатрга узатилади.

1-мисол. А ва В сонларидан каттасини топиш дастури тузилсин. Масала ечимини график равишда қуйидагича тасвирлаш мумкин:



Дастури:

```

10 REM Катта сонни аниқлаш
20 INPUT A, B
30 IF A > B THEN 50
40 PRINT B: GOTO 60
50 PRINT A
60 END

```

Бу дастурни бажараётганда компьютер $A > B$ шартни текширади. Агар шарт бажарилса, у ҳолда 30-сатрдан 50-сатрга ўтилади. Агар шарт бажарилмаса, у ҳолда 40-сатр бажарилади. Шундай қилиб, шартнинг бажарилиши ёки бажарилмаслигига қараб, A ва B сонларининг фақат биттаси – каттаси экранга чиқарилади.

Бундан ташқари, THEN ёрдамчи сўзи ўрнига GOTO ёрдамчи сўзини ишлатиш мумкин. Масалан,

```

30 IF X>0 GOTO 50
40 PRINT X

```

Бу дастурда $X > 0$ шарт текширилади. Агар шарт бажарилса, бошқарув 50-сатрга ўтказилади, акс ҳолда 40-сатр бажарилади.

Бейсикда юқорида қўрилган муносабат шартларидан ташқари қўшимча AND (ва), OR (ёки), NOT (эмас) ёрдамчи сўзларидан тузилган шартларни ҳам ишлатиш мумкин. Масалан, берилган шартларнинг ҳеч бўлмаганда биттасининг бажарилиши натижасида бошқарув 60-сатрга ўтказилиши керак бўлса,

```

10 IF X+Y=0 THEN 60
20 IF Z=A*A THEN 60
30 IF C=10 THEN 60

```

бу операторлар қуйидаги операторга тенг кучли:


```

20 IF X+Y=0 OR Z=A*A OR C=10 THEN 60

```

Шуни таъкидлаб ўтиш керакки, агар текширилаётган шартлар кўп бўлса, улар чапдан ўнга қараб текширилиб боради. Юқоридаги дастур сатрида аввал $X+Y=0$ OR $Z=A*A$ жуфт шартлари текширилади. Сўнгра бу шартлар натижаси билан кейинги шарт текширилади, яъни $(X+Y=0$ OR $Z=A*A)$ OR $C=10$ шартлари текширилади.

2-мисол. X аргументи 1 дан 20 гача 1 қадам билан ўзгарганда $Y=X^2$ функциянинг қийматлари ҳисоблансин. Масала ечимини график равишда қуйидагича тасвирлаш мумкин:



Дастури:


```

10 X=1
20 Y=X^2
30 PRINT X, Y
40 X=X+1
50 IF X ≤ 20 THEN 20
60 END
  
```

Бу дастурда 40-сатр оператори аргумент қийматини 2;3;4;5; тарзда ўзгартириб боради, 50-сатрдаги шарт эса X нинг қиймати 20 дан ошиши ёки ошмаслигини текширади.

3-мисол. Юқоридан эркин тушаётган жисмнинг тезлиги ва баландлигини ҳисоблаш дастурини тузинг. Натижа ҳар 0,1 секунд интервалда жадвал кўринишида чиқарилсин.

Масалани ечиш алгоритминини блок-схемаларда тасвирлаймиз:



Дастури:

```

10 ? "Жисм қандай баландликдан тушяпти" : INPUT X
20 PRINT"_____ "
30PRINT"Тушиш вақти ! Тезлик м/с ! баландлик м
40 PRINT"_____ "
50 G=9.81:T=0
60 H=X-(G*T^2)/2:V=G*T
70 IF H<0 THEN 120
80 PRINT"! ";T;" ";V;" ";H
90 T=T+0.1
100 GOTO 60
120 PRINT"_____ "
    
```

Дастур бажарилгандан кейин дисплей экранида натижалар

жадвал кўринишида берилади (бунда X нинг қиймати 2 га тенг деб олинган):

Жисм қандай баландликдан тушяпти

Тушиш вақти ! Тезлик м/с ! Баландлик м

!	0	0	2
!	.1	.981	1.95095
!	.2	1.962	1.8038
!	.3	2.943	1.55855
!	.4	3.924	1.2152
!	.5	4.905	.77375
!	.6	3.886	.2342

4.9. Цикл оператори.

Цикл дастурнинг шундай бир қисмики, у дастур бажарилиши жараёнида бир неча марта такрорланади. FOR ва NEXT операторлари шундай циклларни ташкил қилиш учун мўлжалланган бўлиб, улар дастур тузишда жуда кўп қулайликларни яратади.

FOR операторини цикл сарлавҳаси оператори деб аталади. Бу операторнинг умумий кўриниши қуйидагича:

m FOR X=A TO B STEP C

Бу ерда:

m – сатр номери;

FOR (учун), TO (гача), STEP (қадам) – хизматчи сўзлар;

X – цикл параметри;

A – цикл параметри X нинг бошланғич қийматини кўрсатадиган арифметик ифода;

B – X нинг юқори (қуйи) чегарасини кўрсатувчи ифода, ўзгарувчи ёки сон;

C – арифметик ифода бўлиб, у X цикл параметрининг ўзгариш қадамини кўрсатади.

Агар C кўрсатилмаган бўлса, у ҳолда унинг қиймати 1 га тенг деб ҳисобланади. У ҳолда FOR оператори қуйидаги кўринишда бўлади:

m FOR X=A TO B

Бошқача айтганда, агар қадам 1 га тенг бўлса, С параметрни ёзиш шарт эмас. FOR операторидан кейин цикл танасини ташкил этадиган операторлар жойлашган бўлади. Ҳар бир цикл NEXT оператори билан тугалланади. Бу операторнинг умумий кўриниши

m NEXT X

бўлиб, бу ерда:

m - сатр номери;

NEXT - хизматчи сўз;

X - цикл параметридир.

FOR оператори бажарилаётганда А арифметик ифоданинг қиймати ҳисобланиб, цикл параметри – X га тенглаштирилади. Сўнгра, одатдагидек FOR дан кейинги операторлар бирин-кетин бажарилади. NEXT операторига дуч келганда цикл параметри С арифметик ифодада кўрсатилган қадамга ошади. (Агар $C < 0$ бўлса, камаяди). X параметр В арифметик ифоданинг қиймати билан таққосланади.

Агар X параметр В дан катта бўлса, у ҳолда цикл тугалланиб, дастурнинг бажарилиши NEXT дан кейинги оператордан бошлаб давом эттирилади.

Агар X параметрнинг қиймати В дан кичик бўлса, у ҳолда бошқарув FOR операторига берилади, FOR ва NEXT операторлари орасидаги операторлар X нинг янги қиймати билан қайтадан бажарилади.

С манфий бўлганда X нинг қиймати камайиб боради ва цикл тугаб, NEXT дан кейинги операторлар бажарилиши учун $X < B$ шарт бажарилиши керак.

1-мисол. Қуйидаги дастурнинг бажарилишини кузатинг:

```
5 FOR X=1 TO 15 STEP 2
10 PRINT "Бу оператор -";X
15 NEXT X
20 PRINT "марта бажарилди"
```

2-мисол. 10 дан 100 гача бўлган 10 га каррали сонларнинг йиғиндисини ҳисобланг.

```
5 S=0
10 FOR I=10 TO 100 STEP 10
```

```

15 S=S+I
20 ? S
25 NEXT I
30 END

```

3-мисол. X аргументнинг 0^0 дан 90^0 гача 10^0 қадам билан ўзгаргандаги қийматлари учун $y=\cos x$ функциянинг қийматлари жадвал кўринишида чиқарилсин. Тригонометрик функцияларнинг қийматлари радианларда ҳисобланишини эслатиб ўтамыз. Градуслардан радианларга ўтиш учун қуйидаги формуладан фойдаланамиз:

радиан = градус * 3.15159/180

Ҳисоблаш дастури қуйидагича:

```

10 REM COS(X) ни ҳисоблаймиз
20 FOR X=0 TO 90 STEP 10
30 G=X*3.15149/180
40 PRINT"X=";X, "Y=";COS(G)
50 NEXT X
60 END

```

Дастур бажарилиши натижасида $\text{Cos}0^0, \text{Cos}10^0, \text{Cos}20^0, \dots, \text{Cos}90^0$ нинг натижалари дисплей экранига жадвал кўринишида чиқарилади:

X=0	Y=1
X=10	Y=.98480777
X=20	Y=.93969272
X=30	Y=.86602562
X=40	Y=.76604482
X=50	Y=.64278817
X=60	Y=.50000076
X=70	Y=.34202111
X=80	Y=.17364933
X=90	Y=1.3267794E-06

FOR ва **NEXT** операторини ишлатиш қодалари.

1. Цикл сарлавҳаси **FOR** ни бажармасдан циклга кириш мумкин эмас. Масалан, қуйида ишлатиш мумкин бўлмаган бошқарувни узатишга мисол келтирилган:

```

10 REM Цикл оператори
.....
50 FOR I=1 TO 5
60 Z=Z+I

```

```
70 NEXT I
80 IF Z < 100 THEN 60
.....
```

Бу ерда 80-сатрдан 60-сатрга, яъни цикл танасига бошқарув узатиляпти. Бу эса мумкин эмас.

2. Исталган пайтда циклдан ташқарига чиқиш мумкин. Масалан:


```
10 FOR X=0.1 TO 0.9 STEP 0.05
20 Y=SIN(X): Z=COS(X)
30 IF Y < Z THEN 90
40 NEXT X
.....
```

3. Цикл танасида А – бошланғич қиймат, В – охири қиймат ва С – қадамнинг қийматларини ўзгартириш тавсия қилинмайди. Бу хатоликка олиб келиши мумкин.

4. Цикл параметрининг охири қиймати цикл тугагач, биттага ошади.

4.10. Мураккаб цикллар.

Цикл операторлари ичма-ич жойлашган бўлиши мумкин. Бир неча бир-бирининг ичида жойлашган циклларни ўзида мужас-самлаштирувчи цикл мураккаб цикл деб аталади. Мураккаб цикллар ички ва ташқи циклларга бўлинади. Масалан,



Бир неча ичма-ич жойлашган FOR операторлари битта сатрда тугалланиши керак бўлса, у ҳолда NEXT операторидан кейин вергул билан ажратилган бир неча цикл параметрларини ёзиш мумкин. NEXT дан кейин ёзилган биринчи цикл параметри энг ички циклга, энг сўнггиси эса ташқи циклга мос келади. Масалан, юқоридаги дастур 100, 110 ва 120-сатрларини қуйидагича битта сатрда ёзиш мумкин:

```
100 NEXT I3, I2, I1
```

Мураккаб циклларни ҳосил қилишда қуйидагиларга эътибор бериш керак:

1. Оддий цикллар учун санаб ўтилган ҳамма қодалар мураккаб цикл учун ҳам тўғри.


2. Ички цикл сарлавҳаси FOR ни бажармасдан ташқи циклдан ички циклга кириш мумкин эмас.

3. Ичма-ич жойлашган цикл параметрлари номлари ҳар хил бўлиши керак.

4. Бирин-кетин жойлашган, бажарилиши тугаган циклларнинг номлари бир хил бўлиши мумкин.

5. Ички цикл бутунлигича ташқи цикл танасига кириши керак.

Қуйида мураккаб циклларни тузиш схемаси келтирилган:



1-мисол. $X=A+B$ ифоданинг $A=3, 5, 7, 9, 11$ ва $B=3, 5, 7$ қийматлари учун 15 та қийматини ҳисобланг.

```

10 FOR A=3 TO 11 STEP 2
20 FOR B=3 TO 7 STEP 2
30 LET X=A+B
35 PRINT A,B,X
40 NEXT B
50 NEXT A

```

Бу дастур бажарилганда дастлаб 10-сатр цикл параметри A га 3 қиймат, 20-сатр ички цикл параметрига ҳам 3 қиймат берилади ва 30-сатрда $X=A+B$ ифоданинг қиймати ҳисобланади. 35-сатрда эса A, B ва ҳисобланган X ўзгарувчиларнинг

қийматлари экранга чиқарилади. 40-сатрдан бошқарув 20-сатрга ўтказилади. Бу ерда В янги 5 қиймат қабул қилади ва 30-сатрда Х янги қийматлар А=3 ва В=5 да ҳисобланади. 40-сатрда эса яна бошқарув 20-сатрга ўтказилади ва В янги қиймат 7 ни қабул қилади, сўнгра яна 30-сатр бажарилади. Шундай қилиб, ички цикл тугалланади ва янги қиймат қабул қилади (А=5). Натижада яна ички цикл бажарилади. Дастур бажарилиши жараёнида ички цикл танаси 15 марта ҳисобланади (5 марта ташқи цикл ва 3*5 марта ички цикл).

2-мисол.

```
10 FOR I=1 TO 9
20 FOR J=1 TO 9
30 ?I;"*";J;"=";I*J: ?
40 NEXT J, I
50 END
```

4.11. DATA, READ, RESTORE операторлари.

Берилган қийматларни киритиш учун DATA-READ операторлари ишлатилиб, уларнинг умумий кўриниши қуйидагича:

```
m DATA a1, a2, ... , aN
. . . . .
n READ x1, x2, ... , xN
```

Бу ерда:

m ва n — сатр номерлари;

DATA — хизматчи сўз бўлиб, "берилганлар" маъносини англатади;

READ — хизматчи сўз бўлиб, "ўқиш" маъносини англатади;

x1, x2, ... , xN — ўзгарувчилар номлари рўйхати бўлиб, уларга мос қийматлар DATA операторида берилган a1, a2, ... , aN ўзгармас катталиклардир.

DATA оператори берилганларни сақлайди. Бу оператор дастурнинг ихтиёрий сатрида сақланади. DATA оператори бажарилмайдиган операторлар синфига мансубдир. DATA операторидаги ўзгарувчилар бир-биридан вергул билан ажратилади. DATA операторида ишлатиладиган ўзгарувчилар ихтиёрий типда бўлиши мумкин. Масалан,

DATA 5, 3.2, -7, "Муסיқа", "Январь"

READ операторидаги ўзгарувчиларнинг типи DATA операторидаги уларга мос келувчи қийматларнинг типига мос бўлиши зарур. Масалан:

```
DATA 5, 3.2, -7, "Муסיқа", "Январь"
READ A, B, C, D$, E$
```

Дастурда бир нечта DATA оператори мавжуд бўлиши мумкин. Лекин улардаги берилганлар ягона берилганлар гуруҳини ташкил қилади. Масалан,

```
5 DATA -7, 22.5, -3, 2
.....
25 DATA 7, 2.9, -8
.....
50 DATA 27, 2.3, -6, -22.5, 0
```

Бу мисолда берилганлар гуруҳи DATA операторларининг дастурда ёзилиши тартибда тузилади. Бу операторлардан тузилган берилганлар гуруҳи ўзида 12 та қийматни қуйидаги тартибда мужассамлаштиради: -7, 22.5, -3, 2, 7, 2.9, -8, 27, 2.3, -6, -22.5, 0. READ оператори навбатдаги ўзгарувчисининг қийматини DATA дан ўқийди.

1-мисол. $A=7$, $B=6$, $C=9$ да Герон формуласи ҳамда DATA-READ операторларидан фойдаланиб, учбурчакнинг юзини ҳисоблаш дастурини тузинг. Берилганларни DATA-READ операторларидан фойдаланиб киритамиз:

```
10 REM Учбурчакнинг юзи
20 DATA 7, 6, 9
30 READ A, B, C
40 P=(A+B+C)/2
50 S=SQR(P*(P-A)*(P-B)*(P-C))
60 PRINT S
```

Бу ҳолда READ операторидан кейин келган A, B ва C ўзгарувчилар мос равишда 7, 6 ва 9 қийматларни қабул қилади. Битта DATA операторига бир неча READ операторлари мурожаат қилиши мумкин. Бундай ҳолни конкрет мисолда кўрамиз:

2-мисол.

```
10 DATA "Москва", "Россиянинг", "Тошкент", "Ўзбекистоннинг"
20 READ A$, B$
30 PRINT A$; " "; B$; " пойтахти"
40 READ A$, B$
50 PRINT A$; " "; B$; " пойтахти"
```

Бу дастур бажарилиши натижасида A\$ ва B\$ ўзгарувчиларга мос равишда "Москва" ва "Россиянинг" қийматлари берилади, сўнгра 30-сатр бажарилгандан кейин дисплей экранида

Москва Россиянинг пойтахти

ёзуви чиқади. 40-сатр бажарилгандан кейин бу ўзгарувчилар мос равишда "Тошкент" ва "Ўзбекистоннинг" қийматларини қабул қилади ва дисплей экранида яна бир ёзув чиқади:

Тошкент Ўзбекистоннинг пойтахти .

3-мисол. 9 ва 10 синфда аъло баҳода ўқийдиган ўқувчилар сони экранга чиқарилсин. Агар 9-а да 12; 9-б да 6; 10-а да 8; 10-б да 13; 10-в да 2 та аълочи борлиги маълум бўлса, у ҳолда дисплей экранига ҳар бир синфда ўқийдиган аълочи ўқувчилар ва жамми аълочи ўқувчилар сони чиқарилсин.

```
10 REM Аълочи ўқувчилар сони
20 DATA 12, 6, 8, 13, 2
30 REM Синф бўйича аълочилар сони
40 READ A1, A2, A3, A4, A5
50 PRINT "9-A"; "9-B"; "10-A"; "10-B"; "10-B"
60 PRINT
70 PRINT " ";A1;" ";A2;" ";A3;" ";A4;" ";A5
80 LET S=A1+A2+A3+A4+A5
90 PRINT "S=";S
```

Дастур бажарилгандан кейин дисплей экранида

```
9-A 9-B 10-A 10-B 10-B
12 6 8 13 2
S=41
```

ни ҳосил қиламиз.

RESTORE оператори берилганлар блокени бошланғич ҳолатга келтириш учун хизмат қилади. Бу операторнинг умумий кўриниши

m RESTORE

Бу ерда m – сатр номери, RESTORE – хизматчи сўз бўлиб, "тиклаш" маъносини англатади.

Бу операторнинг бажарилиши натижасида READ операторидаги ўзгарувчиларга берилганлар блокадаги 1-қийматдан бошлаб қийматлар берилади.

4.12. Массивлар. DIM оператори.

Массив – бу бир ном билан белгиланган қийматлар гуруҳи ёки жадвали. Массивнинг ҳар бир элементи массив номидан сўнг қавс ичига олинган рақам ёки арифметик ифода ёзилиши билан белгиланади.

Ҳар бир массивда индекс сони мавжуд. Бу индекслар сони массивнинг неча ўлчовли эканлигини билдиради. Масалан $V(10)$ бир ўлчовли массивни, $C(1,4)$ эса икки ўлчовли массив эканлигини билдиради.

Бир ўлчовли массивда унинг ҳар бир элементи ўзининг жойлашган ўрнини аниқлайдиган битта индекс билан ёзилади. Икки ўлчовли массив элементи ўзи жойлашган сатр ва устун номерлари ёрдамида белгиланади. Шунинг учун икки ўлчовли массив элементи иккита индекс орқали ёзилади. Ҳар бир массив ўзининг махсус номига эга. Массивнинг номи сифатида лотин ҳарфлари ва араб рақамларидан тузилган, лекин лотин ҳарфидан бошланадиган номлар бўлиши керак. Массивнинг ҳар бир элементини ёзишда массив номидан кейин қавс ичига олинган индексини ҳам ёзиш керак. Индекснинг энг кичик қиймати 0 га тенг.

Масалан $A(10)$ – бир ўлчовли, ҳақиқий сонлардан иборат A массивнинг 10-ўрнида жойлашган элемент эканлигини, $MIN\$(22)$ – MIN ҳарфий катталиклардан тузилган массивнинг 22-элементини, $C\%(5)$ – бутун сонлардан тузилган C массивнинг 5-элементини, $SUM(I)$ – ҳақиқий сонлардан тузилган SUM массивининг I -элементини, $C(2,3)$ – икки ўлчовли ҳақиқий сонлардан тузилган C массивнинг 2-сатри ва 3-устуни кесилган нуқтасидаги элементини билдиради.

Ҳар бир массив ўзининг ўлчамига эга. Шунинг учун у дастурда эълон қилинган бўлиши керак. Бейсик тилида массивларни эълон қилиш DIM оператори орқали амалга оширилади.

Бу операторнинг умумий кўриниши қуйидагича:

m DIM <массив номи>, <массив номи>, . . .

Бу ерда

m – сатр номери;

DIM – операторнинг номи;

массив номи – ихтиёрий типли массив номи бўлиб, ундан кейин қавс ичида шу массивнинг максимал ўлчови келади.

Масалан 10-сатрдаги DIM оператори

```
10 DIM A(50)
```

ҳақиқий сонли 51 та $A(0)$, $A(1)$, $A(2)$, ..., $A(50)$ элементдан иборат бўлган A массивни эълон қилади. 20-сатрдаги DIM оператори

```
20 DIM C$(4,5)
```

ўзида 4 сатр ва 5 та устундан иборат икки ўлчовли $C\$$ массивини эълон қилади. Бу массив элементлари сони 20 та.

DIM оператори дастурда массив элементлари билан ишлайдиган операторлардан олдинроқда ёзилиши керак. DIM операторини дастурнинг бошида ишлатиш мақсадга мувофиқдир. Элементлари сони 11 тадан ошмайдиган массивларни эълон қилиш шарт эмас. Масалан, A(6)$, $MIN(10)$, $B(2, 2)$, ... массивлар.

Массив эълон қилиниши билан, бу массив элементлари учун хотирадан махсус жой ажратилади. Массив элементларини компьютер хотирасига киритиш учун INPUT, LINE INPUT, LET, DATA-READ операторлари ишлатилади. Агар дастурда DIM орқали $A(12)$ массиви эълон қилинган бўлса, у ҳолда унинг элементларини компьютерга киритиш учун INPUT операторини қуйидагича ёзиш мумкин:

```
5 DIM A(12)
15 INPUT A(0), A(1), A(2), ..., A(12)
```

Дастурни ишга туширгандан кейин бу ўзгарувчиларга қийматлар бериш мумкин: ?1, 2, 3, ..., -12, 325

Юқорида эълон қилинган $A(12)$ массив элементлари DATA-READ операторлари ёрдамида

```
5 DIM A(12)
10 DATA 1, 2, 3, ..., -12, 325
90 READ A(0), A(1), ..., A(12)
```

кўринишда берилиши ҳам мумкин. Бу дастур ишга туширилгандан кейин 90-сатрда ёзилган ўзгарувчиларга уларнинг қийматлари 10-сатрдаги DATA операторидан мос равишда олинади.

Агар DATA операторидаги сонлар READ операторидаги ўзгарувчилар сонидан кам бўлса, у ҳолда хатолик ҳақида

маълумот берилади. Юқоридаги мисолда А массив 13 та элементдан иборат. Агар массивнинг элементлари сони кўп бўлса, у ҳолда уни INPUT орқали киритиш қийинчиликларга олиб келади. Бундай ҳолларда цикл оператори ишлатилади.

1-мисол:

```
10 DIM A(15)
20 FOR I=0 TO 15
30 INPUT A(I)
40 NEXT I
```

Дастурнинг бу қисми бажарилганда дисплей экранига 16 марта "?" белгиси чиқарилади. Массивнинг ҳамма элементларини бу пайтда клавиатура орқали киритилади. Бу дастурни DATA-READ орқали ҳам ёзиш мумкин:

```
10 DIM A(15)
20 DATA 1, 3, 5, 6, 1, 2, 56, -2, 10, 1, 5, 8, 2.1, 23, 56
30 FOR I=0 TO 15
40 READ A(I)
50 NEXT I
```

2-мисол. Икки ўлчовли массив элементларини киритиш учун мураккаб цикллардан фойдаланилади.

```
10 DIM A%(3,4)
20 FOR I=1 TO 3
30 FOR J=1 TO 4
40 INPUT A(I,J)
50 NEXT J
60 NEXT I
```

Массив элементларини 1 дан бошлаб ишлатиш кенг тарқалган. Шунинг учун кўп ҳолларда массив элементларини 0 дан бошлаб белгилаш ишлатилмайди.

3-мисол.

```
10 REM КўПАЙТМА
20 DIM A(15)
30 FOR I=1 TO 15
40 INPUT A(I)
50 NEXT I
60 P=1
70 FOR I=1 TO 15
80 P=P*A(I)
90 NEXT I
100 ? P
110 END
```

4-мисол. x нинг қиймати 3 дан 4 гача 0,25 қадам билан ўзгарганда $y=25 \sin(0,4x - 0,7)$ функциянинг қийматлари ҳисобланиб, уларнинг натижаси дисплей экранига чиқарилсин.

```
10 DIM Y(10)
20 I=0
30 FOR X=3 TO 4 STEP 0.25
40 Y(I)=25*SIN(0.4*X - 0.7)
50 PRINT " I=";I;" X=";X;" Y(I)=";Y(I)
60 I=I+1
70 NEXT X
```

Дастур бажарилишида x нинг қиймати 3 дан 4 гача 30-сатр асосида 0,25 қадам билан ўзгаради. 40-сатрда $Y(I)$ нинг қиймати ҳисобланади ва экранга (50-сатрдаги операторга кўра) чиқарилади:

```
I=0 X=3 Y(I)= 11.98563846
I=1 X=3.25 Y(I)= 14.11606183
I=2 X=3.5 Y(I)= 16.10544218
I=3 X=3.75 Y(I)= 17.93390227
I=4 X=4 Y(I)= 19.58317274
```

4.13. GOSUB оператори. Қисм дастур.

Мураккаб дастурларнинг бир нечта жойида бир хил вазифани (масалан, квадрат тенгламани ечишни) бажарувчи операторлар гуруҳини қўллашга тўғри келади ва дастурда бир-бирига айнан ўхшаш бир нечта қисмлар вужудга келади. Компьютер хотирасини ва дастур тузувчининг вақтини тежаш мақсадида ушбу қисмлардаги ўхшаш операторлар бир марта асосий дастурдан ажратиб ёзилади ва унга асосий дастур бажарилиш жараёнида бир неча марта мурожаат қилинади. Бундай кўринишда ёзилган дастур бўлаги "қисм дастур" деб аталади. Қисм дастурга мурожаат қилиш GOSUB оператори орқали амалга оширилади. Бу операторнинг умумий кўриниши

m GOSUB n

бўлиб, бу ерда

m – сатр номери,

GOSUB (GOto SUBroutine – қисм дастурга бориш) – операторнинг номи,

n – қисм дастурнинг биринчи сатри номери.

Қуйидаги мисолни қараб чиқамиз:

```

10 DATA 12, 0.25, 3.15, 7, 6, 12.03, 20, 32, 21
20 READ A, B, C
30 GOSUB 100
40 READ A, B, C
.....

```

30-сатрни бажараётганда компьютер биринчи сатри 100-номер билан бошланувчи қисм дастурни бажаради. Қисм дастур бутунлай бажариб бўлингандан кейин бошқарув GOSUB операторидан сўнг ёзилган операторга, яъни 40-сатрга узатилади ва асосий дастурнинг бажарилиши давом эттирилади. Ҳар бир қисм дастур RETURN оператори билан якунланиши керак. Бу операторнинг умумий кўриниши

k RETURN

бўлиб, бу ерда k – сатр номери.

Бу оператор юқорида айтиб ўтганимиздек, асосий дастурга қайтиб, GOSUB оператордан кейин ёзилган операторга қайтишни таъминлайди. Қисм дастурлар асосий дастурдан кейин жойлаштирилади, бунда асосий дастур END оператори билан тугалланиши шарт, акс ҳолда асосий дастур бажарилиб бўлгач, қисм дастур яна бир бор бажарилади ва хатоликка олиб келади:

```

10 REM Асосий дастур
...
n GOSUB n
...
1000 END

n REM Қисм дастур
...
...
k RETURN

```

1-мисол.

```

10 DATA 12, 0.25, 3.15, 7, 6, 12.03
20 READ A, B
30 GOSUB 100
35 S1=P
40 READ A, B
45 GOSUB 100
50 S2=P
55 READ A, B
60 GOSUB 100
70 S3=P
80 ? S1, S2, S3

```

```
90END
100 P=SQR(A*B)
110 RETURN
```

Бу дастур бажарилганда А, В ўзгарувчиларга мос равишда 12 ва 0,25 қийматлар берилади. Шундан сўнг 30-сатрда қисм дастурга мурожаат амалга оширилади. 100-сатрда А, В сонларининг ўрта геометриги, яъни Р ҳисобланади. 110-сатрда жойлашган RETURN оператори ёрдамида асосий дастурга қайтилади ва бошқарув 35-сатрга берилади. 40-сатр орқали А, В ларга янги қийматлар берилади, 45-сатрда яна қисм дастурга мурожаат этилади ва А, В нинг янги қийматлари учун ўрта геометрик миқдор аниқланади. RETURN оператори 50-сатрга қайтишни таъминлайди, 60-сатрда яна бир бор қисм дастурга мурожаат этилади. 80-сатрда S1, S2 ва S3 ўзгарувчиларнинг қийматлари экранга чиқарилади. 90-сатрда дастур ўз ишини тугатади. Бу оператор (яъни END) бўлмаса, қисм дастур яна бир марта ҳисобланади, бу эса хатоликка олиб келади.

2-мисол. GOSUB оператори ёрдамида $F=\sin(2*X)$, $Z=ABS(F)$ бўлганда $Y1=4*Z^3-4.2*F$ ва $Y2=Z^5-2*F$ функцияларнинг қийматлари ҳисоблансин. Бу ерда X ўзгарувчи 2 дан 12 гача 5 қадам билан Y1 функция учун, 5 дан 13 гача 4 қадам билан Y2 функция учун ўзгаради. Натижалар экранга чиқарилсин.

```
10 REM
20 FOR X=2 TO 12 STEP 5
30 GOSUB 140
40 LET Y1=4*Z^3-4.2*F
50 PRINT "X=";X, "Y1=";Y1
55 PRINT
60 NEXT X
70 PRINT
80 FOR X=5 TO 13 STEP 4
90 GOSUB 140
100 LET Y2=Z^5-2*F
110 PRINT "X=";X, "Y2=";Y2
115 PRINT
120 NEXT X
130 END
140 LET F=SIN(2*X)
150 LET Z=ABS(F)
160 RETURN
```

Дастур бажарилгандан кейин экранда қуйидаги натижалар чиқади:

X=2	Y1=4.91240504
X=7	Y1=-.27220727
X=12	Y1=6.77398756
X=5	Y2=1.13569396
X=9	Y2=1.74084515
X=13	Y2=-1.2672675

4.14. ON GOTO ва ON GOSUB операторлари.

Бошқарувни текширилиши керак бўлган шарт асосида бирор сатрга ўтказиш мумкин. Бу IF оператори билан амалга оширилади. IF оператори ёрдамида битта ёки бир неча шартнинг бажарилиши ёки бажарилмаслигини текшириш мумкин. Лекин, текшириладиган шартлар қанча кўп бўлса, у ҳолда дастурни тузиш, уни тушуниш шу қадар қийин бўлади. Қуйида ихтиёрий бирор сон киритиб, уни 2, 3, 4 даражаларга кўтариш дастурини кўриб чиқайлик.

1-мисол.

```

10 REM Соннинг даражалари
15 INPUT "Сон киритинг";A
20 PRINT "1-A^2"
30 PRINT "2-A^3"
40 PRINT "3-A^4"
50 PRINT "4-тамом"
60 INPUT "Керакли рақамни киритинг (1-4)";B
70 IF B=1 THEN 120
80 IF B=2 THEN 140
90 IF B=3 THEN 160
100 IF B=4 THEN 180
110 PRINT "Нотўғри сон киритилди";GOTO 60
120 PRINT "Квадрат ";A^2
130 GOTO 15
140 PRINT "Куб ";A^3
150 GOTO 15
160 PRINT "Тўртинчи даражаси ";A^4
170 GOTO 15
180 END

```

Бу дастурнинг 60-сатри бажариладиганда бажарилиши керак бўлган амал номерини киритиш лозим. 70-100 сатрларда қайси амал бажарилиши кераклиги текширилиб, бошқарув мос сатрга юборилади ва керакли амал бажарилади. В ўзгарувчининг қийматини текшириш учун 4 та IF-THEN ишлатилапти. Агар

бу текширилаётган шартларнинг бирортаси ҳам бажарилмаса, у ҳолда 110-сатр бажарилади, яъни "Нотўғри сон киритилди" ёзуви пайдо бўлиб, бошқарув қайтадан 60-сатрга узатилади.

Бу мисолда кўп учрайдиган ҳоллардан бири қараб чиқилган, яъни бир ўзгарувчининг қийматига кўра дастурнинг ихтиёрий қисмига мурожаат қилинади.

Бундай операцияларни ON GOTO оператори ёрдамида жуда осон бажариш мумкин. ON GOTO операторини ишлатиш натижасида юқоридаги дастурнинг 70-100 сатрларини битта қатор орқали ёзиш мумкин:

```
70 ON B GOTO 90, 110, 130, 150
```

ва 1-мисол қуйидагича бўлади:

```
10 REM Соннинг даражалари
15 INPUT "Сон киритинг";A
20 PRINT "1-A^2"
30 PRINT "2-A^3"
40 PRINT "3-A^4"
50 PRINT "4-тамом"
60 INPUT "Керакли рақамни киритинг (1-4)";B
70 ON B GOTO 90, 110, 130, 150
80 PRINT "Нотўғри сон киритилди":GOTO 60
90 PRINT "Квадрат ";A^2
100 GOTO 15
110 PRINT "Куб ";A^3
120 GOTO 15
130 PRINT "Туртинчи даражаси ";A^4
140 GOTO 15
150 END
```

70- сатрда В нинг қиймати текширилади. Агар В=1 бўлса, у ҳолда бошқарув 90 сатрга, агар В=2 бўлса, у ҳолда 110-сатрга, В=3 бўлганда 130-сатрга, В=4 да эса 150-сатрга ўтказилади.

В ифоданинг қиймати 1 дан кичик ва 4 дан катта бўлса, у ҳолда ON-GOTO операторидан кейин жойлашган оператор бажарилади.

Умумий ҳолда ON ёрдамчи сўзидан кейин арифметик ифода жойлашади, GOTO ёрдамчи сўзидан кейин вергул билан бир-биридан ажратилган сатр номерлари ёзилади. Бу операторнинг умумий кўриниши

ON ифода GOTO m1, m2, ...

бўлиб, бу ерда:

m – сатр номери;

m_1, m_2, \dots - бошқарув юборилиши керак бўлган сатр номерлари;

"ифода" - бу ихтиёрий арифметик ифода бўлиши мумкин. Бу операторнинг бажарилиши натижасида арифметик ифоданинг қийматининг бутун қисми ҳисоблаб олинади ва бошқарув арифметик ифоданинг бутун қийматига тенг бўлган ўринда турувчи сатрга юборилади. Масалан, бизга C ўзгарувчининг қиймати $10 \leq C < 13$ оралиқда бўлганда, бошқарув 120-сатрга узатилиши керак бўлса, у ҳолда ON-GOTO операторини қуйидагича ёзиш мумкин:

```
ON C-9 GOTO 120, 120, 120
```

2-мисол. Ҳафта кунларини сўз билан ёзувчи дастур тузилсин.

```
10 INPUT "Ҳафта кунининг тартиб номерини киритинг"; N
20 ON N GOTO 40, 60, 80, 100, 120, 140, 160
30 PRINT "Бундай номер мавжуд эмас":GOTO 10
40 PRINT "Душанба"
50 GOTO 170
60 PRINT "Сешанба"
70 GOTO 170
80 PRINT "Чоршанба"
90 GOTO 170
100 PRINT "Пайшанба"
110 GOTO 170
120 PRINT "Жума"
130 GOTO 170
140 PRINT "Шанба"
150 GOTO 170
160 PRINT "Якшанба"
170 END
```

Энди ON GOSUB операторини кўриб чиқамиз. Бу операторнинг умумий кўриниши қуйидагича

```
m ON ифода GOSUB N1, N2, \dots, NK
```

бўлиб, бу ерда

m - сатр номери,
ифода - арифметик ифода,
 N_1, N_2, \dots, N_K - қисм дастурларнинг биринчи сатрлари.

Бу буйруқнинг ON-GOTO буйругидан фарқи шуки, ифоданинг қийматига қараб ON-GOTO бошқарувни битта сатрга юборса, ON-GOSUB қисм дастурга юборади. Бу операторнинг ишлатилишига бир неча мисоллар келтирамиз.

3-мисол. Қуйидаги ифоданинг қийматини ҳисобланг:

$$y = \begin{cases} n! , & \text{агар } a=1 \\ \sum_{i=1}^n i , & \text{агар } a=2 \\ n! + \sum_{i=1}^n i , & \text{агар } a=3 \\ 0 , & \text{акс ҳолда} \end{cases}$$

```

10 REM Y нинг қийматини ҳисоблаш
20 INPUT A, N
25 Y=0
30 ON A GOSUB 100, 200, 300
40 PRINT Y
50 END
100 P=1
110 FOR I=1 TO N
120 P=P*I
130 NEXT I
140 Y=P
150 RETURN
200 S=0
210 FOR I=1 TO N
220 S=S+I
230 NEXT I
240 Y=S
250 RETURN
300 GOSUB 100
310 GOSUB 200
320 Y=P+S
330 RETURN

```

4.15. Ностандарт функция DEF FN.

Бейсикда фойдаланувчи стандарт функциялар орқали ўзи томонидан тузиладиган ностандарт функцияларни ҳам ишлаши мумкин. Бундай функциялар машина хотирасини тежаб, дастурни соддалаштиришга ёрдам беради.

Ностандарт функция DEF оператори ёрдамида берилади. Операторнинг умумий кўриниши

m DEF FN X (A)= < ифода >

бу ерда

m — сатр номери;
 DEF — "аниқлаш" маъносини англатадиган хизматчи сўз,
 FN X — фойдаланувчи томонидан аниқланадиган функция-
 нинг номи;

A — формал параметр.
 FN X даги X ўрнига ихтиёрий латинча ҳарфни ишлатиш
 мумкин. Масалан:

```
10 DEF FN A(X)=X^5-COS(X)
```

1-мисол. Қуйидаги ифодани ҳисоблаш дастури тузилсин:

$$Z = \frac{5(\sin x + \cos 2x) - 3(\sin x + \cos 2x)^2}{\sin x + \cos 2x - 1}$$

Бу мисолда $\sin x + \cos 2x$ ифода бир неча марта ҳисобланади.
 Шунинг учун Y(X) ни

```
Y(X)=SIN(X)+COS(2*X)
```

функция сифатида олиш мумкин. Дастурнинг кўриниши
 қуйидагича бўлади:

```
10 DEF FNY(X)=SIN(X)+COS(2*X)
20 INPUT X
30 Z=(5*FN Y(X)-3*FNY(X)^2)/(FNY(X)-1)
40 PRINT Z
```

30-сатрдаги FN Y функциянинг аргументи ҳақиқий аргумент
 деб аталади. Бу аргументлар ўзгармаслар, ўзгарувчилар (шу
 жумладан, ҳарфий), ифодалар бўлиши мумкин.

Фойдаланувчининг ностандарт функцияси ҳарфий, бутун
 сонли ва ҳақиқий бўлиши мумкин. Ҳарфий ностандарт функция
 \$ белгиси, бутун сонли ностандарт функция эса % белгиси
 билан ажратилиши керак. Бу белгилар ностандарт функция
 номидан кейин қўйилади.

DEF оператори бажарилмайдиган оператор бўлиб, барча
 бажарилмайдиган операторлар каби дастурнинг бошида жой-
 лаштирилиши керак.

2-мисол. Шундай дастур тузилсинки, $y(x) = \sqrt{x^2 + \cos x}$ функ-
 ция аниқлансин ва $x=0; 0,1; 0,2; \dots$; 1 бўлганда унинг
 қийматлари ҳисоблансин.

```
10 DEF FN Y(X)=SQR(X*X+COS(X))
20 FOR X=0 TO 1 STEP 0.1
30 PRINT " "; X , FNY(X)
40 PRINT
50 NEXT X
```

Дастур бажарилгач, экранда қуйидагини ҳосил қиламиз:

0	0
.1	.33141125
.2	.48853795
.3	.62090273
.4	.74122759
.5	.85406413
.6	.96158331
.7	1.0649965
.8	1.1650562
.9	1.2622705
1	1.3570081

4.16. HTAB ва VTAB операторлари.

Бу операторлар курсорни берилган ўринга қўйиб, натижаларни, матнларни, жадвалларни чиқариш учун ишлатилади. Бу операторнинг умумий кўриниши қуйидагича:

m HTAB X : VTAB Y

бу ерда

m - сатр номери;

X - сатрдаги позиция номери (устун номери);

Y - сатр номери.

X ўзгарувчининг қиймати 0 дан 39 гача, Y нинг қиймати эса 0 дан 23 гача бўлиши мумкин. Масалан, экраннинг ўртасида 10-устун, 12-сатрдан бошлаб "Исмингиз нима ?" иборасини чиқаришни кўрамиз.

```
10 HTAB 10:VTAB 12
20 PRINT"Исмингиз нима ?"
```

Юқоридаги дастур ишга туширилгандан кейин дисплей экранини 12-сатрининг 10-позициясидан бошлаб "Исмингиз нима ?" ёзуви чиқади.

5. Ҳарфий катталиклар билан ишлаш

Ҳарфий ўзгарувчилар ЭҲМда матнлар билан ишлаш, ҳисоблаш характериға эға бўлмаған масалаларни ечишда катта имконият ярағади.

Ҳарфий катталиклар учун қўшиш амали мавжуд. Бу амал натижасида икки (ёки ундан кўпроқ) ҳарфий ўзгарувчилар қийматлари биттаға бирлаштирилади.

1-мисол.

```
10 A$="Тошкент"
20 B$="Ўзбекистон республикаси"
30 C$=" пойтахти"
40 ? A$+"-"+B$+C$
```

Бу дастур бажарилиши натижасида экранда қуйидаги матн чиқади:

Тошкент - Ўзбекстон республикаси пойтахти

Қуйида ҳарфий ўзгарувчилар учун баъзи стандарт функциялар билан танишамиз.

1. ASC - қиймати бўш бўлмаған матнли сатрнинг биринчи элементининг ASCII коддини берувчи функция, умумий кўриниши:

ASC("белги")

Компьютерда барча белгилар, ҳарфлар, рақамлар код орқали ифодаланади. Ҳозирги кунда дунё миқёсида ASCII (Ахборотлар билан ўзаро алоқа учун Америка стандарт кодлари) кодларидан фойдаланилади.

ASC("A") нинг (лотинча A) қиймати 65 га, ASC("B") нинг қиймати 66 га, ASC("3") нинг қиймати 50 га, ASC("\$") нинг қиймати 36 га тенгдир.

2. LEN - аргументда мавжуд бўлған ҳарфий катталикнинг белгилари сонини аниқлайди.

1-мисол.

? LEN("Информатика")
RETURN тугмаси босилгач, экранда 11 сони ҳосил бўлади.

2-мисол.

10 INPUT "Сўз киритинг";A\$
20 C=LEN(A\$)
30 PRINT A\$;" сўз ";C;" та белгидан иборат"

3. VAL - ҳарфий катталик сифатида ёзилган сонни (арифметик ифодани) ҳақиқий сонга (арифметик ифодага) ўзгартиради. Умумий кўриниши:

VAL("сон ёки арифметик ифода")

Масалан, VAL("12") нинг қиймати 12 сонига, VAL("35.124") нинг эса 35.124 сонига тенг.

4. STR\$ - аргументдаги сон ёки арифметик ифодани ҳарфий катталик сифатида ёзилган сонга (арифметик ифодага) ўзгартиради. Умумий кўриниши:

STR\$(сон ёки арифметик ифода)

Масалан, STR\$(12.235) нинг қиймати "12.235" га тенгдир.

5. CHR\$ функцияси. Бу функция ASC функциясига тескари функция бўлиб, умумий кўриниши қуйидагича:

CHR\$(N)

Бу ерда N - сон. CHR\$- аргумент сифатида ASCII кодларида ёзилган белгининг коди орқали белгининг ўзини беради. N сони 0 ва 255 сонлари орасида бўлиши керак. Масалан:

? CHR\$(66) - "B" лотин ҳарфини,
? CHR\$(247) - "В" рус ҳарфини беради.

3-мисол. Шундай дастур тузингки, бу дастур берилган код орқали шу кодга мос келувчи белгини экранга чиқарсин.


```

10 INPUT "Кодни киритинг", N
20 IF N>255 THEN 50
30 PRINT N ; " кодга мос келувчи белги "; CHR$(N)
40 GOTO 10
50 PRINT "Код 255 дан ошмаслиги керак !": GOTO 10

```

6. **LEFT\$** функцияси. Бу функция ёрдамида берилган матнли сатрнинг чап томонидан N узунликка эга бўлган қисмини қирқиб олиш мумкин. Бу функциянинг умумий кўриниши қуйидагича:

LEFT\$("матнли сатр", N)

Бу ерда "матнли сатр" - ихтиёрий матнли сатр, N-қирқиб олиниши керак бўлган белгилар сони. N нинг ўзгариш қиймати $0 \leq N \leq 255$ бўлиши керак. Масалан:

LEFT\$("ПРАВЕЦ",4) - "ПРАВ" га,
LEFT\$("ПРАВЕЦ",10) - "ПРАВЕЦ " га тенг.

7. **MID\$** - қирқиш функцияси. Бу функциянинг умумий кўриниши қуйидагича:

MID\$("матнли сатр", M, N)

бу ерда:

"матнли сатр" - ихтиёрий матнли сатр;

M - матнли сатрнинг қирқиб олиниши керак бўлган қисмининг биринчи ҳарфи ўрни;

N - қирқиб олиниши керак бўлган белгилар сони. Бу функция "матнли сатр" нинг M - позициясидаги белгидан бошлаб N та белгини қирқиб олади. N кўрсатилмаса M-белгидан бошлаб, сатр охиригача қирқиб олинади. Масалан:

MID\$("МАТЕМАТИКА", 4, 3) - ЕМА га;
MID\$("МАТЕМАТИКА", 3) - ТЕМАТИКА га;
MID\$("МАТЕМАТИКА", 2, 0) - " " га тенг.

8. **RIGHT\$** функцияси. Бу функция орқали ихтиёрий матнли сатрнинг ўнг томонидан берилган узунликдаги қисмини қирқиб олиш мумкин. Бу функциянинг умумий кўриниши

RIGHT\$("матнли сатр", N)

бўлиб,

"матнли сатр" - ихтиёрий матнли сатр,

N - ўнг томонидан қирқиб олинishi керак бўлган қисм узунлиги.

Бу оператор "матнли сатр" нинг ўнг томонидан N та белги қирқиб олади. Масалан:

```
? RIGHTS("ОЗОДАХОН",3)
```

нинг натижаси ХОН дир.

4-мисол. "МАХЛИЁБОНУ" сўзидан "БОНУ" сўзи қирқиб олинсин.

```
10 A$="МАХЛИЁБОНУ"
```

```
20 B$=RIGHTS(A$,4)
```

```
30 ? B$
```

Иловалар

Бейсик тилининг қўшимча оператор ва функциялари

Қуйида бейсик дастурлаш тилининг китобнинг аввалги бобларида кўрилмаган операторлари ва функцияларининг қисқача тавсифи билан таништирамиз.

1. **CLEAR** оператори. Дастурнинг барча ўзгарувчилари ва массивлари элементлари қийматини 0 га тенглайди. Ҳарфий ўзгарувчилар ва массив элементлари қийматлари ҳам " " - "бўй сатр"га тенглаштирилади. Ёзилиши:

CLEAR

2. **CONT** оператори. **STOP** ва **END** операторлари воситасида ёки **CTRL** ва **C** тугмаларини босиб бажарилиши тўхтатилган дастурни ишга туширади. Дастур кейинги оператордан бошлаб бажарилади. Умумий кўриниши:

CONT

3. **HOME** оператори. Матн ҳолатидаги экранни тозалайди ва курсорни экраннинг чап юқори бурчагига ўрнатади. Умумий кўриниши:

HOME

4. **GET** оператори.

Дастурда ушбу оператор учраганда компьютер клавиатурадан битта ҳарф ёки рақам (умуман ихтиёрий белги) киритилишини кутиб туради. Клавиатура орқали битта ҳарф ёки рақам киритилгач, дастурнинг бажарилиши давом эттирилади, бунда

RETURN тугмаси босилмайди. Киритилган белги экранда акс эттирилмайди. Операторнинг умумий кўриниши қуйидагича

GET A\$

бу ерда A\$ - ҳарфий ўзгарувчи.

Операторда ҳақиқий ўзгарувчини ҳам қўллаш мумкин:

GET A

Бу ҳолда фақат 0 дан 9 гача оралиқдаги битта рақам киритиш мумкин, агар кўп хонали сон киритилса, компьютер фақат биринчи киритилган рақамнигина эътиборга олади. Рақамлардан фарқли белги (ҳарф ёки махсус белги) киритилса, A ўзгарувчининг қиймати 0 га тенглаштирилади - клавиатурадан 0 рақами киритилди деб ҳисобланади.

5. INVERSE, NORMAL, FLASH операторлари.

INVERSE оператори орқали инверс ҳолатига ўтилади. Бу оператор бажарилгач, ахборот экранда оқ фонга қора ранг билан чиқарилади.

FLASH оператори бажарилгач, экранга чиқарилаётган матнли ахборот ёниб-ўчиб туради.

NORMAL оператори эса INVERSE ва FLASH ҳолатларини бекор қилиб, одатдаги "қора фонга оқ белгилар" ҳолатини жорий этади.

6. ONERR GOTO оператори. Дастур бажарилиш жараёнида хатолик вужудга келса, бошқарувни мазкур операторда кўрсатилган сатрга узатади. Одатда дастурчилар дастур тузиш жараёнида, вужудга келган хатолар ҳақида тўлиқ маълумотга эга бўлиш, унинг сабабини аниқлаш ва бартараф этиш мақсадида, дастурга хатони таҳлил этувчи махсус қисм дастур киритадилар, операторда шу қисм дастур бошланувчи сатр номери кўрсатилади. Умумий кўриниши:

ONERR GOTO n

бу ерда n - хатолик вужудга келганда ўтилиши лозим бўлган сатр номери.

7. RESUME оператори. Мазкур оператор хатони таҳлил этувчи қисм дастурдан хатолик вужудга келган сатрга қайтишни таъминлайди. У хатони таҳлил этувчи қисм дастурнинг охириги оператори бўлиши керак. Ёзилиши:

RESUME

8. PEEK функцияси. Адреси аргумент сифатида кўрсатилган хотира катакчасидаги ахборотни ўқиш. Умумий кўриниши:

PEEK (A)

бу ерда A - хотира катакчаси адреси. Функция бажарилиши натижасида олинган сон хотиранинг мазкур катакчасига ёзилган байтнинг ўнлик санок системасидаги кўринишидир.

9. **POKE** оператори. Хотиранинг маълум катакчасига маълумот ёзиш. Умумий кўриниши:

POKE A, B

бу ерда

A - хотира катакчаси адреси;

B - мазкур катакчага ёзилаётган байт, 0-256 оралиқдаги бутун сон бўлиши керак.

10. **POS** функцияси. Экраннинг курсор турган устуни номерини аниқлайди. Умумий кўриниши:

POS (N)

бу ерда N - ихтиёрий бутун сон, унинг қиймати функция натижасига таъсир кўрсатмайди. Функция бажарилиш натижаси 0 дан 39 гача бўлган бутун сондир, 0 экраннинг чап томонидан биринчи устунни билдиради.

11. **SETMOD** оператори. Компьютерни 7 битли ёки 8 битли ҳолатлардан бирига ўтказади. Бу ҳолатларда бир белги мос равишда 7 ёки 8 бит орқали ифодаланади. 7 битли ҳолатда фақат бош ҳарфлар билан, 8 битли ҳолатда бош ва кичик ҳарфлар билан ишлаш имкони мавжуд.

SETMOD 0 - 7 битли ҳолатни жорий этиш,

SETMOD 1 - 8 битли ҳолатни жорий этиш.

12. **SPC** функцияси. Ёзилиши:

SPC (арифметик ифода ёки сон)

Функция курсорни аргумент сифатида келтирилган арифметик ифода ёки соннинг қийматига тенг позицияга (курсорнинг жорий позициясига нисбатан) ўнга суради, бунда курсор ўтган позициялардаги белгилар ўчирилади. Мазкур функция **PRINT** оператори билан биргаликда зарур миқдорда бўш жой ташлаш учун ишлатилади.

13. **SPEED** оператори. Бу оператор ахборотнинг экранга чиқарилиш тезлигини белгилайди, умумий кўриниши

SPEED = N

бу ерда N-бутун сон, $0 \leq N \leq 255$ бўлиши керак, 0-минимал тезлик, 255-максимал тезлик.

Қуйидаги дастур Правец-8 ёзувини экранга ҳар хил тезликда чиқаришни намоён этади.

```
10 FOR X=0 TO 255 STEP 20
20 SPEED=X
30 ?"Правец-8": X
40 NEXT X
```

14. **TAB** функцияси. Бу функция курсорнинг горизонтал позициясини белгилайди. Ёзилиши:

ТАВ (N)

бу ерда N – бутун сон бўлиб, курсорнинг горизонтал позициясини билдиради. Функция бажарилгач, курсор шу ўринга кўчади. PRINT оператори билан биргаликда ахборотни экраннинг исталган жойида чиқариш учун фойдаланилади.

15. TEXT оператори. Бу оператор ёрдамида график ҳолатдан матнли ҳолатга ўтилади.

16. TRACE оператори. Дастур бажарилиши мобайнида бажарилаётган сатр номерини экранга чиқариб туради. Дастур тузиш жараёнида ундаги мавжуд хатоларни аниқлаш учун ишлатилади: экранга чиқарилаётган сатр номерларига кўра дастур бажарилиши тўхтаб қолган сатр аниқланади. Ёзилиши:

TRACE

17. NOTRACE оператори. TRACE операторининг бажарилишини тўхтатади.

Бейсик интерпретатори хабарлари.

Бейсик интерпретатори дастурда хато аниқланганда, у тўғрисидаги хабарни экранга чиқаради, дастур бажарилиши тўхтатилади.

Ҳар бир хато ўз кодига эга бўлиб, у компьютернинг икки махсус адресига ёзилади, экранга чиқарилмайди. Хато тўғрисидаги хабарларнинг умумий кўриниши қуйидагича:

? хато IN сатр номери

Дастурда хато аниқланишидан аввал ONERR GOTO оператори бажарилган бўлса, экранга хато тўғрисидаги хабар чиқарилмайди, хато коди ва хатоли сатр номери хотиранинг махсус адресларига ёзилади. Хато аниқлангач, бошланғич адреси ONERR GOTO операторида кўрсатилган хатони таҳлил этувчи дастур бажарилади.

Қуйида Бейсик интерпретаторининг хато тўғрисидаги хабарлари келтирилган, хабарнинг коди ҳам кўрсатилган.

SINTAX ERROR - синтактик хато. Бу хабар бевосита буйруқда ёки киритилган дастур сатрида синтактик хато борлигини билдиради. Код 16.

NEXT WITHOUT FOR ERROR - дастурдаги бирор NEXT оператори FOR операторисиз ишлатилган. Код 0.

RETURN WITHOUR GOSUB ERROR - дастурда бажарилган

RETURN операторлари сони бажарилган GOSUB операторларининг сонидан кўп. Код 22.

OUT OF DATA ERROR - READ оператори бажарилганда, ундаги ўзгарувчиларга DATA операторларида қийматлар етишмаганини билдиради. Код 42. Масалан:

```
10 DATA 0, 1, 3
20 READ A, B, C, D, E
```

D ва E ўзгарувчилари учун DATA операторида қиймат йўқ. **ILLEGAL QUANTITY ERROR** - оператор ёки функцияда фойдаланилган параметрга (ўзгармас, ўзгарувчи ёки ифода) мумкин бўлмаган қиймат берилган. Код 53. Масалан:

1. LOG(-1)
2. График ҳолатда экран элементи координаталари нотўғри кўрсатилса.

OVERFLOW ERROR - ҳисобланган сон абсолют қиймати ишлатиш мумкин бўлган энг катта сон $1.7E+38$ дан катта. Код 69. Масалан:

```
PRINT 960^24
```

OUT OF MEMORY ERROR - бу хабар Бейсиқда ёзилган дастур компьютер хотирасининг бўш қисмига сифмаганида чиқарилади. Бунинг бир неча сабаби бор:

- дастур жуда узун;
- фойдаланилган ўзгарувчилар сони жуда кўп;
- бир-бирини қамровчи цикллар сони 10 тадан ортиқ;
- бир-бирини ўз ичига олувчи қисм дастурлар сони 24 дан ортиқ;
- бир-бирини ўз ичига олувчи қавслар сони 36 тадан ортиқ;
- LOMEM ва HIMEM қийматлари мумкин бўлган оралиқда эмас.

Хато коди 77.

UNDEF'D STATEMENT ERROR - дастурда йўқ сатрга мурожаат қилиш. Код 90. Масалан:

```
50 GOTO 200
...
100 END
```

BAD SUBSCRIPT ERROR - индексли ўзгарувчининг индекси нотўғри кўрсатилган, массивда бу индексли элемент йўқ. Код 107. Масалан:

```
10 DIM A(5)
...
20 PRINT A(1), A(6)
```

```

]RUN
? BAD SUBSCRIPT ERROR IN 20

```

A массивда A(6) элемент мавжуд эмас.
REDIM'D ARRAY ERROR - ушбу хато DIM операторида ўлчами белгиланган массив ўлчамини қайта белгилашга уринилганда вужудга келади. Коди 120. Мисол:

```
10 DIM A(5,5), B(10)
```

```
...
50 DIM A(8,5)
```

```

]RUN
? REDIM'D ARRAY ERROR IN 50

```

DIVISION BY ZERO ERROR - 0 га бўлиш, бўлувчи қиймати 0 га тенг бўлганда чиқарилади. Коди 133.

TYPE MISMATCH ERROR - ўзгарувчи типи унга берилаётган қийматга мос эмас. Масалан, ҳақиқий ўзгарувчига ҳарфий ўзгармас қиймат қилиб берилган: A="САЛОМ". Коди 163.

SIRING TOO LONG ERROR - бу хабар матнли сатрнинг (ҳарфий катталиқнинг) узунлиги 255 дан ортиқ бўлганда чиқарилади. Коди 176.

FORMULA TOO COMPLEX ERROR - IF-THEN операторида шарт сифатида ҳарфий катталиқларни таққослаш амали ҳам ишлатилиши мумкин. Агар дастурда 2 тадан ортиқ шундай IF-THEN операторлари бажарилса, ушбу хато вужудга келади. Коди 191.

INDEF'D FUNCTION ERROR - аниқланмаган фойдаланувчи функциясидан фойдаланиш. Коди 224.

REENTER - бу хабар клавиатурадан киритилган қиймат INPUT операторидаги ўзгарувчи типига мос келмаганда ёки қиймат киритилмай RETURN клавиши босилганда чиқарилади.

Қиймат тўғри киритилгандан сўнг, программа бажарилиши давом этади. Коди 254.

BREAK IN . . . - бу хабар дастурнинг STOP оператори ёрдамида ёки CTRL ва C тугмаларини босиш билан тўхтатилганлигини билдиради. Дастур бажарилиши қайси сатрда тўхтатилгани кўрсатилади, масалан:

```
BREAK IN 90.
```

```
Коди 255.
```

ILLEGAL DIRECT ERROR - бу хабар INPUT, GET ва DEFFN операторлари бевосита тартибда берилганда чиқарилади. Бу операторлардан фақат дастурда фойдаланиш мумкин. Бу хабарнинг коди йўқ.

Мундарижа

Сўз боши	3
1. Асосий тушунчалар	4
1.1. Бейсик тилининг асосий элементлари	4
1.2. Берилганларни ифодалаш	5
1.3. Стандарт математик функциялар	6
1.4. Арифметик ифодалар	8
2. Бевосита ҳисоблаш тартиби	9
3. Бейсик тилининг асосий буйруқлари	12
4. Бейсик тилининг асосий операторлари	16
4.1. Изоҳ оператори - REM	17
4.2. LET - ўзлаштириш оператори	18
4.3. PRINT оператори	19
4.4. INPUT оператори	23
4.5. LINE INPUT оператори	27
4.6. STOP ва END операторлари	28
4.7. Шартсиз ўтиш оператори GOTO	29
4.8. Шартли ўтиш оператори IF - THEN	30
4.9. Цикл оператори	35
4.10. Мураккаб цикллар	38
4.11. DATA, READ ва RESTORE операторлари	40
4.12. Массивлар. DIM оператори	43
4.13. GOSUB оператори. Қисм дастур	46
4.14. ON GOTO ва ON GOSUB операторлари	49
4.15. Ностандарт функция DEF FN	52
4.16. HTAB ва VTAB операторлари	54
5. Ҳарфий катталиклар билан ишлаш	55
Иловалар	58
Бейсик тилининг қўшимча оператор ва функциялари	58
Бейсик интерпретатори хабарлари.	61